HEWLETT-PACKARD

Step-by-Step Solutions For Your HP Calculator Electrical Engineering





Electrical Engineering

Step-by-Step Solutions for Your HP-42S Calculator



Edition 1 November 1988 Reorder Number 00042-90021

Notice

This book and any keystroke programs contained herein are provided "as is" and are subject to change without notice. Hewlett-Packard Company makes no warranty of any kind with regard to this book or the keystroke programs contained herein, including, but not limited to, the implied warranties of merchantability and fitness for a particular purpose. Hewlett-Packard Company shall not be liable for any errors or for incidental or consequential damages in connection with the furnishing, performance, or use of this book or the keystroke programs contained herein.

• Hewlett-Packard Company 1988. All rights reserved. Reproduction, adaptation, or translation of this book, including any programs, is prohibited without prior written permission by Hewlett-Packard Company; except as allowed under the copyright laws. Hewlett-Packard Company grants you the right to use any program contained in this book in a Hewlett-Packard calculator.

The programs that control your calculator are copyrighted and all rights are reserved. Reproduction, adaptation, or translation of those programs without prior written permission of Hewlett-Packard Company is also prohibited.

Corvallis Division 1000 N.E. Circle Blvd. Corvallis, OR 97330, U.S.A.

Printing History

Edition 1

November 1988

Mfg. No. 00042-90023

Contents

5 How To Use This	s Book
-------------------	--------

1	7 8 13 17 23 28 32	Circuit Calculations Voltage Division ("V÷") Current Division ("I÷") Power Triangle ("PWR3") Frequency Response of Transfer Function ("FQRS") RC Timing ("RC") Delta-Wye Conversions ("DY")
2	38	Network Analysis
	39	Using the Circuit Editor ("CIRCT")
	39	Starting the "CIRCT" Program
	40	Initializing the Circuit
	40	Displaying Circuit Elements
	41	Adding Circuit Elements
	42	Deleting a Circuit Element
	42	Printing the Circuit
	43	Quitting "CIRCT"
	43	Saving Circuits
	43	Storing Elements
	53	Mesh Analysis ("MESH")
	63	Nodal Analysis ("NODAL")
	67	Impedance of a Ladder Network ("LADDR")

3	72 73 82	Filter Design Active Filter Design ("AF") Butterworth Filter Design ("BF")
4	95 96 108	Transmission Lines Transmission Line Calculations ("LINE") Transmission Line Impedance ("TLI")
5	115 116 123	Amplifier Analysis Transistor Amplifier Performance ("TAP") Transistor Amplifier Bias Optimization ("BIAS")
6	136 137 139 143	Truth Tables Writing a Logical Expression as a Program Printing a Truth Table ("PTTBL") An Interactive Truth Table ("ITTBL")
7	147 148 149 149 150 156 157 158 159	Utilities Circuit Calculation Utilities Impedance of an Element ("C→Z" and "L→Z") Combining Parallel Impedances ("ZP") Entering Radian Frequency ("FQ?") Circuit Element Input Utility ("EL?") Function Name Utility ("FCN?") Yes/No Utility ("Y?N") Product Over Sum Utility ("P/S") Size Utility ("SIZE?")

How To Use This Book

Please take a moment to familiarize yourself with the formats used in this book.

Organization. Each chapter in this book covers a different area of electrical engineering. Sections within each chapter highlight the use of each program. The sections are organized like this:

- Description of the program, including equations and variables used.
- Special remarks and limitations.
- General instructions.
- Keystroke examples.
- Program listings.

About the Examples. Unless otherwise stated, the keystrokes and displays shown in each section assume the following conditions:

- The required programs have been keyed into the calculator.
- The stack is clear and you're using the specified display format. Generally, this does not affect the results of the example, but your displays may not exactly match the ones in this book.
- The SIZE is set to 25 registers (the default). The number of registers needed (if any) is listed under "Remarks."

As you work the examples, remember that lowercase letters are displayed as uppercase letters when they appear in menu labels. **If You Have a Printer.** Many of the programs in this book will produce printed output if printing is enabled. Press **PRINT A PON** to enable printing.

If you are not using a printer, be sure to disable printing (**PRINT A FOFF**) to avoid losing results.

About Program Listings. It is assumed that you understand how to key programs into your calculator. If you're not sure, review part 2, "Programming," in the owner's manual.

If you print your programs, remember that the printer may print some characters differently than they are displayed. (For example, the \downarrow character is printed as \lor .) Also note that some printers cannot print the angle character (\checkmark).

About the Subject Matter. Discussions on the various topics included are beyond the scope of this book. Refer to basic texts on the subjects of interest. Many references are available in university libraries and in technical and college bookstores. The examples in this book demonstrate approaches to solving problems, but they do not cover the many ways to approach general problems in electrical engineering.

Our thanks to Dex Smith of TwentyEighth Street Publishing for developing this book.

1

Circuit Calculations

This chapter contains programs that solve for an unknown circuit parameter (when the other parameters are known), define a power triangle from voltage and current, and convert impedances between delta and wye circuit configurations.

Voltage Division ("V÷")

For a circuit in the following general form, the "V÷" program solves for any of the four complex values provided the other three are known.



Variables Used.

In Equation	Description	In Program
V _T	Terminal voltage (volts).	VT
V ₂	Voltage across impedance Z_2 .	V2
Z_1	Impedance (ohms).	Z1
Z_2	Impedance (ohms).	Z2

Since any of these values can be a complex number, the Solver cannot be used. The following program ("V÷") emulates the Solver by displaying a menu containing the four variables in the above equation.

Remarks.

- "V÷" does not alter the angular and coordinate modes; you may use them as you wish when keying in complex values.
- Flag 21 (printer enable) is set or cleared to match flag 55 (printer existence). This automatically produces printer output if flag 55 is set.

Program Instructions.

- 1. Key the "V÷" program (listed on page 10) into your calculator.
- **2.** Press **XEQ** $\forall \div$ (to run the "V \div " program).
- **3.** Use the variable menu displayed by the program to store the known values.
- 4. Press the key for the value you want to calculate.
- 5. To work another problem, go to step 3; to quit, press EXIT.

Example. Given the following circuit and voltage measurements, what must the impedance, Z_2 , be?



Select Degrees and Polar modes, select the FIX 2 display format, and run the "V÷" program.

MODE	S DEG	MO	DES	POLAR
DISP	FIX	02 [XEQ	- P-	

×:0.	00			
VT	V2	Z1	22	

Store the known values.

80 ENTER 10 COMPLEX

VT=8	30.00	3 ∡1	0.00	
٧T	72	Z1	22	

15 Z1

70 ENTER 30 COMPLEX V2

Now, solve for the unknown	n impedance.
----------------------------	--------------

22

The unknown impedance (Z_2) is 37.71 \measuredangle 79.29° ohms.

"V÷" Program Listing.

00 (148-Byte Prom)

Program:

Defines menu variables.

Comments:

Z1=15.00

Sets or clears flag 21 to match flag 55.

Displays the variable menu and stops.

VT V2 21 22

V2=7	0.00	2 ∡3	0.00)
٧T	47 2	21	22	

Z2=37.71 ∡79.29 VT V2 21 22

01 LBL "V+" 02 MVAR "VT" 03 MVAR "V2" 04 MVAR "Z1" 05 MVAR "Z2" 06 CF 21 07 FS? 55 08 SF 21 09 LBL 00 10 CLA 11 VARMENU "V+" 12 STOP

Determines which variable was 13 ATOX 14 ATOX selected by subtracting the ASCII 15 codes of the first two characters in the variable name. For example, 16 XEQ IND ST X when you press Z1 to calculate 17 GTO 00 Z1, the program branches to LBL 41 because the ASCII code of "Z" is 90, the ASCII code of "1" is 49, and 90 - 49 = 41.Calculates V2. 18 LBL 36 19 RCL "Z2" 20 RCL× "VT" 21 RCL "Z1" 22 LASTX 23 + 24 ÷ 25 STO "V2" 26 VIEW "V2" 27 RTN Calculates VT. 28 LBL 02 29 RCL "Z1" 30 RCL+ "Z2" 31 RCL× "V2" 32 RCL÷ "Z2" 33 STO "VT" 34 VIEW "VT" 35 RTN Calculates Z2. 36 LBL 40 37 RCL "V2" 38 RCL× "Z1" 39 RCL "VT" 40 LASTX 41 -42 ÷ 43 STO "Z2" 44 VIEW "Z2" 45 RTN

46 LBL 41 47 RCL "Z2" 48 RCL× "VT" 49 RCL÷ "V2" 50 RCL- "Z2" 51 STO "Z1" 52 VIEW "Z1" 53 END Calculates Z1.

Current Division ("I÷")

For a circuit in the following general form, the "I÷" program solves for any of the four complex values provided the other three are known.



Variables Used.

In Equation	Description	In Program
IT	Terminal current (amps).	I.T
I ₂	Current in impedance Z_2 .	12
Z_1	Impedance ₁ (ohms).	Z1
Z_2	Impedance ₂ (ohms).	Z2

Since any of these values can be a complex number, the Solver cannot be used. The following program ("I÷") emulates the Solver by displaying a menu containing the four variables in the above equation.

Remarks.

- "I÷" does not alter the angular and coordinate modes. You may use them as you wish when keying in complex values.
- Flag 21 (printer enable) is set or cleared to match flag 55 (printer existence). This automatically produces printer output if flag 55 is set.

Program Instructions.

- 1. Key the "I÷" program (listed on page 15) into your calculator.
- **2.** Press XEQ I÷ (to run the "I÷" program).
- **3.** Use the variable menu displayed by the program to store the known values.
- 4. Press the key corresponding to the value you want to calculate.
- 5. To work another problem, go to step 3; to quit, press EXIT.

Example. Given the following circuit, what is the input current (I_T) if the current (I_2) through the 10 Ω resistor is 12 $\not\propto$ 45° amperes?



Select Degrees and Polar modes, select the FIX 2 display format, and run the I÷ program.

MODES	DEG	MODES	POLAR	×: 0.	00			
DISP	FIX 02	XEQ I	-	<u> </u>	12	Z1	22	

Key in the three known values.

17.5 Z 1

Z1=1	7.5	3		
I.T	15	21	22	

10 Z2

12 ENTER 45 COMPLEX I2

Now calculate the unknown cur	rent

I.T

The unknown current (I_T) is 18.86 \measuredangle .45° amperes.

"I+" Program Listing.

Program:

00 (155-Byte Prom) 01 LBL "I÷" 02 MVAR "I.T" 03 MVAR "I2" 04 MVAR "Z1" 05 MVAR "Z2" 06 CF 21 07 FS? 55 55. 08 SF 21 09 LBL 00 10 CLA 11 VARMENU "I÷" 12 STOP 13 ATOX 14 ATOX 15 -16 XEQ IND ST X 17 GTO 00

Defines menu variables.

Comments:

Sets or clears flag 21 to match flag 55.

Displays the variable menu and stops.

Determines the selected variable by subtracting the ASCII codes of the first two characters in the variable name. For example, when you press I.T., the program branches to LBL 27 because the ASCII code of "I" is 73, the ASCII code of "." is 46, and 73 - 46 = 27.

I.T=	18.8	86 ∡	45.0	0	
I.T	12	Z1	22		

Z2=1	0.0	3		
1.T	12	21	22	

I2=12.00 ∡45.00

18 19 20 21 22 23	LBL 23 RCL "Z1" RCL× "I.T" RCL "Z2" LASTX +
24	÷
25	STO "I2"
26	VIEW "I2"
27	RIN
28 29 30	LBL 27 RCL "Z2" RCL+ "Z1"
31	RCL× "I2"
32	RCL÷ "Z1"
33	STO "I.T"
34	VIEW "I.T"
35	RIN
36 37 38 39 40 41 42 43	LBL 40 RCL "Z1" RCL× "I.T" RCL÷ "I2" RCL- "Z1" STO "Z2" VIEW "Z2" RTN
4.4	
44	LDL 41 RCL 4124
46	RCL × "72"
47	RCL "I.T"
48	LASTX
49	-
50	÷
51	STO "Z1"
52	VIEW "Z1"
53	END

Calculates I2.

Calculates I.T.

Calculates Z2.

Calculates Z1.

Power Triangle ("PWR3")

The "PWR3" program calculates any value for the power triangle, provided that certain other values are known.



Variables Used.

In Figure	Description	In Program
V	Voltage (volts).	٧
Ι	Current (amperes).	I
Р	Average power (watts).	Р
Q	Reactive power (vars).	Q
S	Apparent power (watts).	s
	Power factor ($\cos \theta$).	Pf

Remarks.

- Be sure to enter V and I as complex numbers. If the values in V and I are not complex numbers, the program will stop and display Invalid Type. If you generate this error, *restart* the program by pressing <u>EXIT</u> <u>XEQ</u> <u>PWR3</u>.
- The "PWR3" program sets Degrees and Polar modes.

- The "PWR3" program uses flag 10 to control the calculations of V and I.
- A minus sign preceding a (result or input) value for power factor indicates a *lagging* power factor.

Program Instructions.

- 1. Key the "PWR3" program (listed on page 20) into your calculator.
- **2.** Press XEQ PWR3 (to run the "PWR3" program).
- **3.** Use the variable menu displayed by the program to store the known values and to calculate the unknowns:

To calculate P, Q, S, or pf:

- **a.** Key in the voltage (in *polar* form) and press
- **b.** Key in the current (in *polar* form) and press **I**.
- C. Calculate any of the four values by pressing P, Q, S, or PF.

To calculate the voltage, V:

- **a.** Key in the current (in *polar* form) and press I.
- **b.** Key in the apparent power and press **S**.
- **c.** Key in the power factor and press **PF**.
- **d.** Calculate the voltage by pressing

To calculate the current, *I*:

- **a.** Key in the voltage (in *polar* form) and press
- **b.** Key in the apparent power and press **S**.
- **c.** Key in the power factor and press **PF**.
- **d.** Calculate the current by pressing **I**.
- 4. To work another problem, go to step 3; to quit, press EXIT.

Example. For a circuit with an applied voltage of $100 \ge 10^{\circ}$ volts and a resulting current of 2.85 \measuredangle - 40° amperes, determine the power triangle and the power factor.

Select the FIX 2 display format, and run the "PWR3" program.

DISP FIX 02 (XEQ) PWR3	X: 0.00 V I P Q S PF
	V=100.00 ∡10.00 V I P Q S PF
2.85 (ENTER) 40 (+/-) COMPLEX I	I=2.85 ∡-40.00 V I P Q S PF
P	P=183.19 V I P & S PF
The average power is 183.19 watts.	
Q	Q=218.32 V I P Q S PF
The reactive power is 218.32 vars. (The s indicates if Q is a leading or a lagging va	sign of the power factor (Pf) alue.)

1

The apparent power is 285 watts.

PF

The power factor is 0.64. Because a minus sign indicates a lagging power factor, the reactive power, Q, is also lagging. If all other variables remain unchanged, what voltage would be required to increase the apparent power to 300 watts?

300 S

S=30	0.0	0			
Ŷ		P	Q	S	PF

I P Q S PF

P Q S

PF

S=285.00

pf=-0.64

P



The required voltage is 105.26 ightarrow 10.00° volts.

"PWR3" Program Listing.

Comments:

00	(193-Byte Prgm)	
01	LBL "PWR3"	Declares the menu variables.
02	MVAR "V"	
03	MVAR "I"	
04	MVAR "P"	
05	MVAR "Q"	
Ø6	MVAR "S"	
07	MVAR "pf"	
08	DEG	Sets Degrees and Polar modes. Sets
09	POLAR	or clears flag 21 to match flag 55.
10	CF 21	6
11	FS? 55	
12	SF 21	
13	LBL A	Displays variable menu and stops.
14	VARMENU "PWR3"	Pressing R/S redisplays the menu.
15	CLA	
16	STOP	
17	ATOX	
18	X=0?	
19	GTO A	
20	ХТОА	The ASCII character code of the
21	1	first letter in the variable name is
22	AROT	subtracted from 73 to determine
23	R₊	which routine to execute.
24	73	
25	-	
26	XEQ IND ST X	

27 ASTO ST L Stores the variable name into the Last X-register and displays the 28 STO IND ST L 29 VIEW IND ST L result. 30 GTO A 31 LBL 13 Calculates V. 32 RCL "I" 33 SF 00 34 XEQ B 35 RTN 36 LBL 00 Calculates I. 37 RCL "V" 38 CF 00 39 XEQ B 40 RTN 41 LBL 07 Calculates P. 42 XEQ 39 43 RCL "V" 44 RCL× "I" 45 ABS 46 RCL× "pf" 47 ABS 48 RTN Calculates Q. 49 LBL 08 50 XEQ 39 51 XEQ C 52 RCL "pf" 53 ACOS 54 SIN 55 × 56 RTN Calculates S. 57 LBL 10 58 XEQ C 59 ABS 60 RTN

61 LBL B Calculates V if flag 00 is set, or I if 62 COMPLEX flag 00 is clear. 63 RCL "pf" 64 FS? 00 65 +/-66 ACOS 67 + 68 ENTER 69 ENTER 70 RCL "S" 71 R+ 72 ÷ 73 X<>Y 74 COMPLEX 75 FC?C 00 76 +/-77 RTN 78 LBL C Calculates the magnitude of VI. 79 RCL "I" 80 RCL× "V" 81 ABS 82 RTN 83 LBL 39 Calculates pf. 84 RCL "V" 85 COMPLEX 86 RCL "I" 87 COMPLEX 88 X<>Y 89 R+ 90 -91 COS 92 LASTX 93 SIGN 94 +/-95 X 96 STO "pf" 97 END

Frequency Response of Transfer Function ("FQRS")

For a transfer function of the form

$$G(S) = \frac{K(Z_1S + 1)}{S^N(Z_2S + 1)(Z_3S + 1)\left(\frac{S^2}{\omega_0^2} + \frac{2Z_4S}{\omega_0} + 1\right)}$$

the "FQRS" program calculates G(S) and $\log |G(S)|$ for any input frequency ω (where $S = j\omega$).

Variables Used.

In Equation	Description	In Program
K	Transfer function parameter.	к
N	N Transfer function parameter.	
Z ₁ Transfer function parameter.		Z1
Z ₂ Transfer function parameter.		Z2
	Transfer function parameter.	Z3
Z_4	Transfer function parameter.	Z4
ω_0	Transfer function parameter.	ωØ
ω	Input frequency, $2\pi f$ (radians/sec).	ω

Remarks.

- For type 0 systems, enter N = 0.
- Z_1, Z_2 , and Z_3 can be entered as 0. If there is no quadratic term, enter Z_4 as 0 and ω_0 very large compared to $1/Z_3$, where Z_3 is the smallest first-order term used (other than zero).
- The "FQRS" program sets Degrees and Polar modes.

Program Instructions.

- 1. Key the "FQRS" program (listed on page 26) into your calculator.
- **2.** Press **XEQ FORS** (to run the "FQRS" program).
- **3.** The program prompts you for values of K, N, Z_1 , Z_2 , Z_3 , Z_4 , and ω_0 . At each prompt, key in the value and press [R/S].
- **4.** The program displays G(S) and $\log |G(S)|$ and stops. Press **R/S** to go to step 3 for another problem.

Example. Find G(S) and $\log |G(S)|$ for

$$G(S) = \frac{12(S + 0.6)}{S(S + 1)(S^2 + 6S + 36)}$$

The frequency, ω , is 0.01 rad/sec.

First put G (S) into proper form:

$$G(S) = \frac{.2(1.67S + 1)}{S(S + 1) \left[\left(\frac{S}{6} \right)^2 + \left(\frac{S}{6} \right) + 1 \right]}$$

Select the FIX 2 display format and run the "FQRS" program.

DISP FIX 02 (XEQ) FORS	Y: 0.00 K?0.00
.2 (R/S)	Y: 0.20 N?0.00
1 [<u>R/S</u>]	Y: 1.00 Z1?0.00

1.67 [R/S]

Y:1.67 Z2?0.00	

1 <u>R/S</u>	Y: 1.00 Z3?0.00
0 (<u>R/S</u>)	Y: 0.00 Z4?0.00
.5 (<mark>R/S</mark>)	Y: 0.50 w0?0.00
6 (<u>R/S</u>)	Y: 6.00 w?0.00
.01 (<u>R/S</u>)	G(S)=20.00 ∡-89.71 LOG IG(S)I=1.30

G(S) and log |G(S)| for the given conditions are 20.00 \measuredangle -89.71° and 1.30.

"FQRS" Program Listing.

Program:

Comments:

00 (159-Byte Prom) 01 LBL "FQRS" 02 INPUT "K" 03 INPUT "N" 04 INPUT "Z1" 05 INPUT "Z2" 06 INPUT "Z3" 07 INPUT "Z4" 08 INPUT "w0" 09 LBL 00 10 INPUT "w" 11 DEG 12 RECT 13 1 14 X<>Y 15 RCL÷ "w0" 16 X+2 17 2 18 RCL× "Z4" 19 RCL× "w" 20 RCL÷ "w0" 21 + 22 COMPLEX 23 1 24 RCL "Z3" 25 RCL× "w" 26 COMPLEX 27 X 28 1 29 RCL "Z2" 30 RCL× "w" 31 COMPLEX 32 X 33 POLAR 34 COMPLEX 35 90

Prompts for each input value.

Calculates G(S).

36 RCL× "N" 37 + 38 +/-39 X<>Y 40 RCL "w" 41 RCL "N" 42 Y+X 43 X 44 17X 45 X<>Y 46 COMPLEX 47 RECT 48 RCL "K" 49 ENTER 50 RCL× "Z1" 51 RCL× "w" 52 COMPLEX 53 × 54 POLAR 55 "G(S)=" 56 ARCL ST X 57 ABS 58 LOG 59 ⊢"⊑LOG |G(S)|=" 60 ARCL ST X 61 AVIEW 62 END

Displays results.

RC Timing ("RC")

The "RC" Solver program computes any of the six variables in the following figure and equation, provided the other five are known.



$$V_{i} = V_{1}e^{(-t/RC)} + V_{2}\left(1 - e^{(-t/RC)}\right)$$

By rearranging terms, the equation can be written as:

$$\left[\frac{tC}{\left[-e\left(1-\frac{V_i-V_1}{V_2-V_1}\right)\right]}\right] - R = 0.$$

where e is the base of natural logarithms.

Variables Used.

In Equations	Description	In Program
V ₁	Voltage before step (volts).	٧1
V ₂	Voltage after step (volts).	V2
V_i	Instantaneous voltage (volts).	Vi
R	Resistance (ohms).	R
С	Capacitance (farads).	С
t	Time (seconds).	t

Remarks.

- The "RC" Solver program uses only *real*-number inputs.
- For voltages across the resistor and capacitor, $V_R + V_C = V$ applies *at all times*.

Program Instructions.

- **1.** Key the "RC" Solver program (listed on page 31) into your calculator.
- 2. Select the "RC" Solver program: press SOLVER RC.
- **3.** Use the variable menu to store the five known variables: key in the value and press the corresponding key.
- **4.** Press the key corresponding to the unknown variable. The Solver searches for the unknown and displays the solution (if one can be found).
- **5.** To work another problem, go to step 3, or press **EXIT EXIT** to quit.

Example. A 555 type of integrated circuit timer uses an external RC configuration for time determination. When used as a one-shot, its output pulse terminates when the capacitor charges to two-thirds of the supply voltage. Until the pulse starts, the capacitor is shorted so V1 = 0. Given a supply voltage of 12V and a 47 μ F capacitor, what size resistor should you use to generate a one-second pulse?

Select the ENG 3 display format and the "RC" Solver program.

DISP ENG 03 SOLVER RC	X: 0.000E0 V1 V2 V1 R C T
0 V1	V1=0.000E0 V1 V2 VI R C T
12 V2	V2=12.00E0 V1 V2 VI R C T
2 ENTER 3 ÷ 12 🗙 🛛 🛛 I	Vi=8.000E0 V1 V2 VI R C T
47 E 6 🖅 C	C=47.00E-6 W1 W2 W1 R C T
1 <u>T</u>	t=1.000E0 V1 V2 VI R C T

R

R=19.37E3 V1 V2 VI R C T

Use a 19.37 Kohm resistor.

"RC" Program Listing.

```
00 ( 58-Byte Prgm )
                       12 RCL "V2"
01 LBL "RC"
                        13 LASTX
02 MVAR "V1"
                        14 -
03 MVAR "V2"
                        15 ÷
04 MVAR "Vi"
                        16 -
05 MVAR "R"
                        17 LN
06 MVAR "C"
                        18 +/-
07 MVAR "t"
                        19 RCL× "C"
Ø8 1
                        20 RCL÷ "t"
09 RCL "Vi"
                        21 1/X
10 RCL "V1"
                        22 RCL- "R"
11 -
                        23 END
```

Delta-Wye Conversions ("DY")

This program allows you to convert impedance values between delta and wye configurations. That is, given the three wye values $(Z_{Y1}, Z_{Y2}, \text{ and } Z_{Y3})$, you can calculate any of the three delta values $(Z_{12}, Z_{23}, \text{ and } Z_{13})$. Likewise, given the delta values, you can calculate any of the wye values.



Variables Used.

In Figure	Description	In Program
Z_{Y1}	Impedance (ohms).	ZY1
Z_{Y2}	Impedance (ohms).	ZY2
Z _{Y3}	Impedance (ohms).	ZYS
Z ₁₂	Impedance (ohms).	Z12
Z ₁₃	Impedance (ohms).	Z13
Z ₂₃	Impedance (ohms).	Z23

Remarks.

- Flag 21 (printer enable) is set or cleared to match flag 55 (printer existence). This automatically produces printer output if flag 55 is set.
- The program will give erroneous results (original inputs) if the *calculated* outputs are used directly as inputs.

Program Instructions.

- 1. Key the "DY" program (listed on page 35) into your calculator.
- Select the coordinate mode you want to use: press ■MODES
 RECT for Rectangular mode, or ■MODES
 POLAR for Polar mode.
- 3. Press XEQ DY to run the "DY" program.
- **4.** The program displays a variable menu containing the six variables in the above illustration.

To calculate wye values,

- **a.** Store each of the three delta values by keying in the value and pressing the corresponding menu key.
- **b.** Press the key for the wye value you want to calculate. (Repeat for each of the other two unknown wye values.)

To calculate delta values,

- **a.** Store each of the three wye values by keying in the value and pressing the corresponding menu key.
- **b.** Press the key for the delta value you want to calculate. (Repeat for each of the other two unknown delta values.)
- 5. You can work as many problems as you want. The menu stays active until you press **EXIT** or select an application menu.

Example. Given the following delta circuit, determine the equivalent wye circuit.



Select the Rectangular mode and the FIX 4 display format, and run the "DY" program.

■MODES RECT ■DISP FIX 04 (XEQ) DY	X: 0.000 271 272 273 212 223 213
Enter the delta values.	
45 Z13	213=45.0000 271 272 275 212 225 213
0 ENTER 22 COMPLEX Z12	Z12=0.0000 i22.0000 271 272 275 212 225 213
5 ENTER 17 +/ COMPLEX Z23	Z23=5.0000 -i17.0000 271 272 273 212 223 213
Now calculate each of the wye values.	
ZY1	ZY1=1.9604 i19.6040 zy1 zyz zyz z12 zz3 z13
ZY2=7.6238 i1.4376

ZYS



"DY" Program Listing.

Э

Program:

00	< 219-Byte Prom
01	LBL "DY"
02	MVAR "ZY1"
03	MVAR "ZY2"
04	MVAR "ZY3"
05	MVAR "Z12"
Ø6	MVAR "Z23"
07	MVAR "Z13"
as	CE 21
аq.	5, 21 FS2 55
10	CE 21
10	JF ZI
11	LBL A
12	VARMENU "DY"
13	STOP
14	ATOX
15	XTOA
16	ATOX
17	XTOA
18	ATOX
19	XTOA
20	+
21	95
22	-
23	XED IND ST X
<u> </u>	new the of n

Comments:

Declares menu variables.

Sets or clears flag 21 to match flag 55.

Returns the ASCII character codes of the selected variable name and restores the name in the Alpha register.

Adds the ASCII codes of the last two characters of the selected variable and subtracts 95 to determine which subroutine to execute. 24 SF 25 25 ÷ 26 ASTO ST L 27 FC?C 25 28 XEQ 03 29 STO IND ST L 30 FC? 50 31 VIEW IND ST L 32 GTO A 33 LBL 43 34 RCL "Z12" 35 RCL× "Z13" 36 GTO 01 37 LBL 44 38 RCL "Z12" 39 GTO 00 40 LBL 45 41 RCL "Z13" 42 LBL 00 43 RCL× "Z23" 44 LBL 01 45 RCL "Z12" 46 RCL+ "Z13" 47 RCL+ "Z23" 48 RTN 49 LBL 04 50 RCL "ZY3" 51 GTO 02 52 LBL 05 53 RCL "ZY2" 54 GTO 02 55 LBL 06 56 RCL "ZY1"

Stores the variable name in the Last X register and displays the result. Division by zero is detected with flag 25 (*error ignore*) and an appropriate message is displayed. If this happens, line 31 is skipped because flag 50 (*message*) is set.

Calculates Z_{Y1} .

Calculates Z_{Y2} .

Calculates Z_{Y3} .

Shared subroutines.

Calculates Z_{12} .

Calculates Z_{13} .

Calculates Z_{23} .

Shared subroutine. 57 LBL 02 58 RCL "ZY1" 59 RCL× "ZY2" 60 LASTX 61 RCL× "ZY3" 62 + 63 RCL "ZY2" 64 RCL× "ZY3" 65 + 66 X<>Y 67 RTN Division by zero indicates an open circuit (infinite resistance). Infinity is approximated with 9.99×10^{499} . 68 LBL 03 69 ⊢"=Open Circuit" 70 9.99⊧499 71 AVIEW 72 END

Network Analysis

This chapter contains programs for performing mesh and nodal network analyses for circuits containing any combination of resistors, capacitors, inductors, and general impedances. Circuits for mesh analysis may also contain voltage sources and circuits for nodal analysis may contain current sources.

Circuits are entered into the calculator using the "CIRCT" program and then processed by running the "MESH" or "NODAL" program.

Using the Circuit Editor ("CIRCT")

"CIRCT" is a menu-driven program that allows you to add, delete, and print elements in a circuit. The circuit elements are stored in a matrix named *CIRCT*, where each row in the matrix contains an element.

Required Programs. "CIRCT" (page 45), "EL?" (page 151), and "Y?N" (page 157).

Starting the "CIRCT" Program

If there are no elements stored in *CIRCT* (if no circuit exists), you'll see this display when you run the "CIRCT" program:

XEQ CIRCT

{ 0-Element Circuit }

Or, if the matrix CIRCT exists, the program assumes that it contains n circuit elements (where n is the number of rows in the matrix), and you'll see a display like this:

XEQ CIRCT

{ 7-Element Circuit } MOD DEL

You can view the elements in the circuit by using the \blacktriangle and \bigtriangledown keys.



If the *CIRCT* matrix exists but does not contain circuit information that the "CIRCT" program recognizes, the program can be thrown far off course. To prevent this from happening, either clear the *CIRCT* variable (CLEAR)

CLV CIRCT) before starting the program, or initialize the circuit as shown in the next section.

Initializing the Circuit

To initialize *CIRCT* (that is, to delete the *CIRCT* matrix), press INIT YES.

INIT

INITIAL	IZE	Circ	ui t'	?
YES				ND

YES

{ 0-Element Circuit }

When there are no circuit elements, note that the DEL, PRINT, and INIT menu keys are *not* displayed or active.

Displaying Circuit Elements

Use the \blacktriangle and \bigtriangledown keys to move up and down through the list of elements.

Here's a typical display:



For nodal analysis this display reads: "The third circuit element is a voltage source connected between node 2 and node 4 with a value of $120.00 \leq 60.00^{\circ}$ volts." (For mesh analysis, the location numbers would represent *meshes* 2 and 4.)

When a circuit element is displayed, the menu keys **RDD**, **DEL**, **PRINT**, and **INIT** are active even though they are not visible. (If you press • to clear the message, you'll see that the menu *is* active.)

Adding Circuit Elements

- 1. Use the ▲ and ▼ keys to move to the position in the list of circuit elements where you want to add an element. The new element will be inserted *after* the displayed element.
- Press ADD, the *first* menu key. (The menu keys are active, even though the message temporarily covers them.) The program displays Location: # [ENTER] # and a menu of the six types of circuit elements.
- 3. Key in the location of the element.

For mesh analysis: mesh# [ENTER] mesh#, where the two mesh numbers indicate the meshes that share the element. For an element in a single mesh, press mesh# [ENTER].

For nodal analysis: from-node# [ENTER] to-node#, where the circuit element is connected from the first node to the second.

The order of the location numbers is important only when entering a voltage or current source. For example, the illustration on page 40 shows a voltage *rise* from node 2 to node 4. If the node numbers had been entered in the opposite order, the value should be negative to represent a voltage *drop* from node 4 to node 2.

- 4. Press a menu key to specify the type of element you want to add:
 - R (resistor, ohms)

C (capacitor, farads)

(inductor, henrys)

Z (general impedance, ohms)

₩, (voltage source, volts)

I (current source, amperes)

5. The program displays a menu of the common units for the element you're entering. Key in the value of the element and then press the appropriate key. (For example, to enter a 1,000-ohm resistor you could press 1000 OHM or 1 KOHM.)

All elements are converted to their default units (see the table on page 44) before being added to *CIRCT*.

To add another element, go to step 3. To return to the main menu, press **EXIT**.

Examples later in this chapter demonstrate how to use the "CIRCT" program for entering circuit elements for mesh and nodal analyses.

Deleting a Circuit Element

To delete a circuit element:

- **1.** Use the \blacktriangle and \bigtriangledown keys to display the element you want to delete.
- 2. Press DEL, the second menu key. (The menu keys are active, even though the message temporarily covers them.)
- 3. To prevent accidentally deleting an element, the program displays DELETE Element? Press YES to complete the operation. (Press NO or EXIT) if you change your mind.)

Printing the Circuit

If you want a printed record of the circuit, press **PRINT**, the fifth menu key. Note that this printing is slower than most other printing operations because it involves printing graphics in the display.

The following sample output was printed after entering the circuit on page 54:

(7-Element Circuit)
(1):1=0*1
150.00 &10.00
(2):1 --- 1
16.00
(3):1--- 2
0.00 i20.00
(4):1--- 3
0.00 -i24.00
(5):2 --- 2
32.00
(6):2 --- 3
12.00
(7):3--- 3
0.00 -i16.00

Quitting "CIRCT"

Press $\boxed{\text{EXIT}}$ to exit the "CIRCT" program. After exiting, if you don't run any other programs (or move the program pointer in any other way), you can restart "CIRCT" by pressing $\boxed{\text{R/S}}$.

Saving Circuits

To save a copy of your circuit:

- 1. Press EXIT to quit "CIRCT".
- **2.** Press **RCL** CIRCT to recall a copy of *CIRCT*.
- **3.** Press **STO ENTER** *name* **ENTER** to store a copy of *CIRCT* (where *name* is the name of a new variable).
- 4. Press **R/S** to restart the "CIRCT" program.

To restore a saved copy of a circuit:

- 1. Press EXIT to quit "CIRCT".
- 2. Press RCL name to recall a copy of a circuit that's been saved.
- **3.** Press **STO CIRCT** to store a copy of the matrix into *CIRCT*.
- **4.** Press **R/S** to restart the "CIRCT" program.

Storing Elements

The *CIRCT* variable contains a three-column matrix. Each row in the matrix represents a single circuit element:

- The first column contains a number in the form *tt.mmnn* where *tt* is the type code for the element (see the table below), *mm* is the first location number, and *nn* is the second location number. If *tt.mmnn* is negative, the value for the element is complex.
- The second column contains the *real* value of the element in the default units (see the table below). Voltage and current sources are stored in Polar form; all other elements are stored in Rectangular form.
- The third column contains the *imaginary* part of the element value.

Element Type	Units	Type Code
Resistor	Ohms	82
Capacitor	Farads	67
Inductor	Henrys	76
General impedance	Ohms	90*
Voltage source	Volts	86*
Current source	Amperes	73*
*If the element value is a complex numbe	r, the type code is neg	ative.

Remarks.

- CIRCT is an n x 3 matrix containing the circuit elements. An example of entering the elements of a circuit is contained in the next section, "Mesh Analysis."
- Values for all elements are displayed using the default units (see the table above).
- Flag 09 is set when the editor is extracting a complex value from *CIRCT*.
- Flag 10 is set if there is no variable *CIRCT*.
- Flag 25 is used to determine if *CIRCT* exists.
- Register R_{00} contains the current circuit-element number.
- Register R₀₁ contains the total number of circuit elements.

"CIRCT" Program Listing.

Program:

Comments:

00 (626-Byte Prom) 01 LBL "CIRCT" 02 PROFF 03 CLX 04 STO 00 05 STO 01 06 CF 10 07 SF 25 08 INDEX "CIRCT" 09 FC?C 25 10 SF 10 11 FS? 10 12 GTO A 13 RCL "CIRCT" 14 DIM? 15 X<>Y 16 STO 01 17 LBL A 18 CLMENU 19 "ADD" 20 KEY 1 GTO 01 21 "DEL" 22 FC? 10 23 KEY 2 GTO 02 24 "PRINT" 25 FC? 10 26 KEY 5 XEQ 12 27 "INIT" 28 FC? 10 29 KEY 6 GTO 06 30 KEY 7 XEQ 07 31 KEY 8 XEQ 08

32 KEY 9 GTO 99

Initializes the program.

Defines the menu for the program. If flag 10 is set (no circuit), defines only the "ADD" key.

```
Displays the menu and stops.
33 LBL B
34 XEQ 00
35 MENU
36 STOP
37 GTO B
                               Inputs a circuit element using the
38 LBL 01
                               "EL?" utility on page 151.
39 XEQ "EL?"
40 FC?C 08
41 GTO A
42 FS? 10
                               Inserts the new circuit element into
43 1
                               the CIRCT matrix.
44 FS?C 10
45 GTO 03
46 RCL 00
47 X=0?
48 GTO 10
49 RCL 01
50 X≠Y?
51 GTO 05
52 R.
53 R.
54 RCL "CIRCT"
55 DIM?
56 R¥
57 1
58 +
                               Increases (or creates) the CIRCT
59 LBL 03
60 3
                               matrix for the new element.
61 DIM "CIRCT"
62 INDEX "CIRCT"
                               Stores the new circuit element into
63 LBL 04
64 CLX
                               CIRCT.
65 1
66 STOIJ
67 R.
68 R¥
```

69 X<>Y 70 ÷ 71 X<>Y 72 REAL? 73 GTO 11 74 COMPLEX 75 X<>Y 76 → 77 X<>Y 78 LBL 11 79 STOEL 80 RCLIJ 81 X<>Y 82 STO 00 83 1 84 STO+ 01 85 GTO 01 86 LBL 05 87 I+ 88 INSR 89 R¥ 90 R+ 91 RCLIJ 92 GTO 04 93 LBL 10 94 R+ 95 1 96 ENTER 97 STOIJ 98 INSR 99 GTO 04

Inserts a new row into the *CIRCT* matrix.

Inserts a new row at the top of the *CIRCT* matrix.

100 LBL 00 101 RCL 00 102 IP 103 X=0? 104 GTO C 105 FS? 10 106 GTO C 107 "(" 108 AIP 110 1 111 STOIJ 112 RCLEL 113 CF 09 114 XK0? 115 SF 09 116 ABS 117 IP 118 LASTX 119 FP 120 100 121 × 122 IP 123 AIP 124 LASTX 125 FP 126 100 127 × 128 ⊢" п 129 AIP 130 ⊢"'' 131 J+ 132 R+ 133 R+ 134 RECT 135 86 136 X=Y? 137 POLAR 138 R+

Displays the current element. If $R_{00} = 0$, then displays top-of-list message. (Note that line 128 contains three blank spaces between the double quotes.)

139 73 140 X=Y? 141 POLAR 142 R+ 143 RCLEL 144 ENTER 145 FS? 09 146 → 147 FS?C 09 148 COMPLEX 149 ARCL ST X 150 AVIEW 151 REAL? 152 R+ 153 R+ 154 1 155 ENTER 156 32 157 POSA 158 6 159 × 160 4 161 + 162 XEQ IND ST T 163 AGRAPH 164 RTN 165 LBL C Displays top-of-list message. 166 RCL 01 167 "C " 168 AIP 169 ⊢"-Element " 170 ⊢"Circuit)" 171 AVIEW 172 RTN 173 LBL 82 Alpha string to produce graphic 174 "**¿* *¿* *** resistor. 175 RTN

Alpha string to produce graphic 176 LBL 67 capacitor. 177 "++++|+|++++" 178 RTN Alpha strings to produce graphic 179 LBL 76 inductor. 180 "++E" 181 4 182 XTOA 183 ⊢"⊑" 184 XTOA 185 ⊢"E++" 186 R+ 187 RTN Alpha string to produce graphic 188 LBL 90 189 "++8(((((8++" impedence. 190 RTN 191 LBL 86 Alpha string to produce graphic voltage source. 192 "fff8DDD8f&f" 193 RTN 194 LBL 73 Alpha string to produce graphic 195 "↔+8DDD8+T8+" current source. 196 RTN 197 LBL 02 Deletes the current circuit element. 198 RCL 00 199 X=0? 200 GTO A 201 "DELETE Element?" 202 XEQ "Y?N" 203 X=0? 204 GTO A 205 SF 25 206 DELR 207 FC?C 25 208 GTO 14 209 RCLIJ 210 X<>Y

211 STO 00 212 1 213 STO- 01 214 GTO A 215 LBL 06 216 "INITIALIZE " 217 F"Circuit?" 218 XEQ "Y?N" 219 X=0? 220 GTO A 221 LBL 14 222 CLV "CIRCT" 223 CLX 224 STO 00 225 STO 01 226 SF 10 227 GTO A 228 LBL 07 229 RCL 00 230 1 231 -232 XK0? 233 RCL 01 234 STO 00 235 RTN 236 LBL 08 237 1 238 STO+ 00 239 RCL 01 240 RCL 00 241 X>Y? 242 CLX 243 STO 00 244 RTN

Initializes the pointers and deletes *CIRCT* variable.

Decrements the element pointer.

Increments the element pointer.

245 LBL 12 246 RCL 01 247 1E-3 248 × 249 STO 00 250 LBL 13 251 XEQ 00 252 PRON 253 PRLCD 254 CLA 255 PRA 256 PROFF 257 ISG 00 258 GTO 13 259 CLX 260 STO 00 261 RTN 262 LBL 99 263 EXITALL 264 END

Prints the graphic display for circuit elements in *CIRCT*.

Exits all menus.

Mesh Analysis ("MESH")

The "MESH" program (listed on page 57) calculates the mesh currents of a circuit containing any combination of resistors, capacitors, inductors, general impedances, and voltage sources. (If the circuit you want to analyze contains *current* sources, convert them to voltage sources.)

How "MESH" Works. "MESH" uses the elements in the *CIRCT* matrix to create the following matrices:

$$\mathbf{I} = \frac{\mathbf{Z}}{\mathbf{V}} = \frac{MATA}{MATB} = MATX$$

Variables Used.

In Equation	Description	In Program
	$n \times 3$ matrix containing circuit elements.	CIRCT
Z	Impedance matrix, Z.	МАТА
V	Voltage matrix, V.	МАТВ
I	Solutions matrix.	МАТХ
	Radian frequency, $2\pi f$ (radians/sec).	ω

Since the variable names *MATA*, *MATB*, and *MATX* are used, you can use the Simultaneous Equations application to work with the data after using "MESH". Press MATRIX SIMO nn (where nn is the number of mesh currents).

Remarks.

- Flag 08 is cleared for mesh analysis.
- Flag 10 is set when the location numbers are equal (for mesh analysis).
- "MESH" leaves the calculator in the Polar mode.
- Register R₀₀ contains the element counter. Register R₀₂ contains the number of mesh currents. Register R₀₃ is used for intermediate

results. Be sure to set the SIZE to at least four registers (■MODES) ▼ SIZE 4 ENTER) before running "MESH".

Programs Required. "MESH" (page 57), "CIRCT" (page 45), "C→Z" (page 148), "EL?" (page 151), "FQ?" (page 149), and "Y?N" (page 157).

Program Instructions.

- 1. Enter the circuit elements using the "CIRCT" program. Press **EXIT** when you are finished.
- If you want the results to be printed, press PRINT A PON to enable printing. If you're not using a printer, be sure to disable printing (press PRINT A POFF).
- **3.** Press **XEQ MESH** (to run the "MESH" program).
- When you see No. Mesh Currents?, key in the number of mesh currents and press (R/S).
- 5. When you see Radian Frequency(2πf)?, key in the radian frequency and press R/S. (This value is used only if the circuit contains capacitance or inductance.)
- 6. The mesh currents are then calculated and displayed. If you're not using a printer, press R/S after each result is displayed.

Example: Calculating Mesh Currents. Use the "CIRCT" program to enter the following circuit. Then execute "MESH" to calculate the mesh currents, I_1 , I_2 , and I_3 .



First, select FIX 2 display format and then enter the circuit elements using the "CIRCT" program.

DISP FIX 02 [XEQ] CIRCT

If the circuit contains any elements, delete them by initializing it.

INIT YES

Now, add the new elements to the circuit.

ADD

Enter all of the elements in mesh 1. Since the voltage source isn't shared by another mesh, its location is entered like this:

1 ENTER V

Enter the value for the voltage source.

150 [ENTER] 10 [COMPLEX] V

Enter the location and value for the 16-ohm resistor. Like the voltage source, the resistor is unique to mesh 1, so its location is entered the same way (1 [ENTER]).

1 [ENTER] R 16 OHM

Since a complex impedance is provided for the inductor, enter it as a general impedance. This element is shared between mesh 1 and mesh 2, so enter both location numbers.

1 [ENTER] 2 Z 0 [ENTER] 20 COMPLEX OHM

Enter the capacitor shared by mesh 1 and mesh 3 in the same way.

1	EN	TER]	3 Z	0	(ENTER)	24
+	<u>/_</u>	[CO	MPLEX)	Ol	-111	

{ 0-Element Circuit

> ocation: # [ENTER] LZ

7-Element Circuit

ADD DEL

PRINT INIT





ocation: # [ENTER] # LZY



Value?

					,	
_		 	 	 		
-						



.ocation: # [ENTER] # C L Z V I Now enter the elements in mesh 2 that have not already been entered.

2 [ENTER] R 32 OHM

Loca	atio	ר: #	[Eh	ITER:] # [
6	C	L	Z	Ŷ	

2 [ENTER] 3 R 12 OHM

Loca	atio	n: #	[Eh	ITER] #
R	C	L	Z	Ŷ	1

Location: # [ENTER] RCLZV

Enter the last element, which is the capacitor unique to mesh 3. (Note that you could have entered these elements in any order.)

3	ENTER	B Z	0	ENTER 16	
+	/_] []	COMPLEX		OHM	

Press **EXIT** to return to the main level. You'll see the last element that you entered.

EXIT

At this point you can use \blacktriangle and \bigtriangledown to view the elements in the circuit.

Exit from the "CIRCT" program and calculate the mesh currents. If you want the results printed, press **PRINT A PON** to enable printing.

EXIT

XEQ MESH

No. Mesh Currents? x:0.00

Key in the number of mesh currents.

3 [R/S]

Radian Frequency(2πf)? x:0.01

Since the inductor and capacitors were given as complex impedances, the frequency will not be used. Press [R/S] to display the current in mesh 1.

R/S

I1 =x: 6.28 ∡4.72

(7):3--3 0.00 -i16.00

Y: 1.00 X: 34.00

If you're not using a printer, press \mathbb{R}/\mathbb{S} to display the current in mesh 2.

R/S

I2= ×:2.61 ∡53.37

If you're not using a printer, press \mathbb{R}/\mathbb{S} again to display the current in mesh 3.

R/S

I3= ×:3.09 ∡-2.73

"MESH" and "NODAL" Program Listing.

Program:

Comments:

Clears flag 08 (to indicate mesh analysis) and prompts for the number of mesh currents.

Sets flag 08 (to indicate nodal analysis) and prompts for the number of nodes.

Stores the number of mesh currents or nodes.

Creates the complex matrices *MATA* and *MATB*, and prompts for a frequency.

00 01 02 03 04 05	<pre>(491-Byte Prom LBL "MESH" CF 08 "No. Mesh " H"Currents?" GTO A</pre>	>
06 07 08	LBL "NODAL" SF 08 "No. Nodes?"	
09 10 11 12	LBL A RCL 02 PROMPT STO 02	
13 14 15 16 17 18 20 21 22 23 24 25	ENTER NEWMAT ENTER COMPLEX STO "MATA" DIM? 1 STO ØØ NEWMAT ENTER COMPLEX STO "MATB" XEQ "FQ?"	

```
26 CF 21
27 "Working..."
28 AVIEW
29 FS? 55
30 SF 21
31 LBL a
32 INDEX "CIRCT"
33 RCL 00
34 1
35 SF 25
36 STOIJ
37 FC?C 25
38 GTO D
39 RCLEL
40 XEQ IND ST X
41 1
42 STO+ 00
43 GTO a
44 LBL B
45 FS? 08
46 1ZX
47 ENTER
48 R+
49 R.
50 INDEX "MATA"
51 STOIJ
52 RCLEL
53 R+
54 FS? 10
55 GTO b
56 -
57 STOFL
58 R.
59 X<>Y
60 STOIJ
61 R+
```

Displays a "working" message and sets or clears flag 21 to match flag 55.

Sets the index pointer to the current element (identified in R_{00}). If this generates an error, then the end of the *CIRCT* matrix has been reached.

Recalls the element type code and executes the appropriate subroutine to add the element value to *MATA* and *MATB*.

Increments the element pointer and repeats the loop for the next element.

Accumulates an element value into *MATA*. If flag 10 is set (indicating an element shared by two meshes or connected to the reference node), adds the value only to the matrix element on the main diagonal. If flag 10 is clear, the value is accumulated to the appropriate matrix elements not on the diagonal. Note that values accumulated along the diagonal are positive; values accumulated in other matrix elements of *MATA* are negative.

62 64 65 66 67 68 69 70 72	STOEL CLX LASTX R↑ STO 03 R↑ ENTER STOIJ R↓ RCLEL R↑	
73 74 75 76 77 78	XEQ 6 RCL 03 ENTER STOIJ RCLEL LASTX	
79 80 81 82	LBL b + STOEL RTN	Stores a value into the matrix.
83 84 85 86	LBL 82 XEQ F XEQ e GTO B	Stores a resistor value.
87 88 89 90 91	LBL 67 XEQ F XEQ e XEQ "C→Z" GTO B	Stores a capacitor value.
92 93 94 95 96	LBL 76 XEQ F XEQ e XEQ "L→Z" GTO B	Stores an inductor value.
97 98	LBL 90 XEQ F	Stores a general impedance value.

```
99 XEQ e
100 XEQ e
101 RECT
102 COMPLEX
103 GTO B
104 LBL 73
105 LBL 86
106 XEQ F
107 XEQ e
108 XEQ e
109 POLAR
110 COMPLEX
111 INDEX "MATB"
112 RCL ST Z
113 FC? 08
114 GTO c
115 FC? 10
116 GTO c
117 X≠0?
118 GTO C
119 X<> ST Z
120 X<>Y
121 -1
122 ×
123 X<>Y
124 LBL C
125 1
126 STOLJ
127 RCLEL
128 R+
129 +
130 STOEL
131 RTN
132 LBL c
133 1
134 STOLJ
135 CLX
136 RCL ST T
```

Stores a current or voltage source into *MATB*. Since the two types are not distinguished, all sources are assumed to be voltage sources for mesh analysis or current sources for nodal analysis.

Adds a current source connected to the reference node.

Adds a voltage source (mesh analysis) or a current source *not* connected to the reference node (nodal analysis). 137 ENTER 138 RCLEL 139 R+ 140 + 141 STOEL 142 R+ 143 X=Y? 144 RTN 145 1 146 STOIJ 147 RCLEL 148 LASTX 149 -150 STOEL 151 RTN 152 LBL D 153 POLAR 154 RCL "MATB" 155 RCL÷ "MATA" 156 STO "MATX" 157 SF 21 158 INDEX "MATX" 159 LBL d 160 RCLIJ 161 R+ 162 "I" 163 FS? 08 164 "V" 165 AIP 166 ⊢"=" 167 RCLEL 168 FS? 55 169 ARCL ST X 170 AVIEW 171 I+ 172 FC? 77 173 GTO d 174 RTN

Calculates the results using matrix division.

Prepares to display the results.

Recalls each result from *MATX* and displays it with the appropriate label. If a printer is being used (flag 55 set), the value is appended to the label.

Flag 77 is tested to determine when the end of *MATX* has been reached.

```
175 LBL e
176 J+
177 RCLEL
178 RTN
179 LBL F
180 ABS
181 FP
182 100
183 ×
184 IP
185 LASTX
186 FP
187 100
188 ×
189 CF 10
190 X=Y?
191 SF 10
192 FC? 08
193 RTN
194 FS? 10
195 GTO J
196 XEQ G
197 XEQ G
198 X<>Y
199 FC? 10
200 RTN
291 +
202 ENTER
203 RTN
204 LBL G
205 X=0?
206 SF 10
207 X<>Y
208 RTN
209 LBL J
210 "Invalid Data"
211 PROMPT
212 END
```

Increments the column pointer and recalls a matrix element.

Decodes the element location into two mesh or node numbers in the Xand Y-registers.

Sets flag 10 to indicate that the two location numbers are the same.

For nodal analysis (flag 08 set), displays an error message if the location numbers are equal. (The numbers do not make sense if they're equal.)

Sets flag 10 if the reference node is used.

Error message.

Nodal Analysis ("NODAL")

The "NODAL" program (listed on page 57) calculates the node voltages of a circuit containing any combination of resistors, capacitors, inductors, general impedances, and current sources. (If the circuit you want to analyze contains *voltage* sources, convert them to current sources.)

How "NODAL" Works. "NODAL" uses the circuit information in the *CIRCT* matrix to create the following matrices:

$$\mathbf{V} = \frac{\mathbf{Z}}{\mathbf{I}} = \frac{MATA}{MATB} = MATX$$

Variables Used.

In Equation	Description	In Program
	$n \times 3$ matrix containing circuit elements.	CIRCT
Z	Impedance matrix.	МАТА
v	Voltage matrix.	МАТХ
I	Current matrix.	МАТВ
	Radian frequency, $2\pi f$ (radians/sec).	ω

Since the variable names *MATA*, *MATB*, and *MATX* are used, you can use the Simultaneous Equations application to work with the data after you've finished using "NODAL". Press \blacksquare MATRIX \blacksquare SIMO nn (where nn is the number of nodes).

Remarks.

- Flag 08 is set for nodal analysis.
- Flag 10 is set when one location number is zero (for nodal analysis).
- "NODAL" leaves the calculator in Polar mode.
- Register R₀₀ contains the element counter. Register R₀₂ contains the number of nodes. Register R₀₃ is used for intermediate results. Be

sure to set the SIZE to at least four registers (MODES) SIZE 4 ENTER) before running "NODAL".

Programs Required. "MESH" (page 57), "CIRCT" (page 45), "C→Z" (page 148), "EL?" (page 151), "FQ?" (page 149), "Y?N" (page 157).

Program Instructions.

- 1. Enter the circuit elements using the "CIRCT" program. Press **EXIT** when you're finished.
- If you want the results to be printed, press PRINT A FON to enable printing. If you're *not* using a printer, be sure to press
 PRINT A FOFF to disable printing.
- **3.** Press **XEQ** NODAL (to run the "NODAL" program).
- 4. When you see No. Nodes?, key in the number of nodes and press **R/S**.
- 5. When you see Radian Frequency(2πf)?, key in the radian frequency and press R/S. (This value is used only if the circuit contains capacitance or inductance.)
- 6. The node voltages are then calculated and displayed. If you're not using a printer, press **R/S** after each result is displayed.

Example: Calculating Node Voltages. Use the "CIRCT" program to enter the following circuit. Then execute "NODAL" to calculate the node voltages, V_1 , V_2 , and V_3 .



2: Nodal Analysis 65

DISP FIX 02 (XEQ) CIRCT { 7-Element Circuit 3

If "CIRCT" has any elements stored in it, delete them by initializing it.

Begin by selecting FIX 2 display format and running "CIRCT".

INIT YES

Now, add the new elements to the circuit.

ADD

Enter the current source from node 0 to node 1.

0 [ENTER] 1 I 3 [ENTER] 35 COMPLEX A

Enter the resistor from node 1 to node 2.

1 [ENTER] 2 R 8 0HM

Enter the impedance from node 0 to node 2.

0 [ENTER] 2 Z 4 [ENTER] 9

Enter the resistor from node 0 to node 3.

0 [ENTER] 3 R 17 0HM

Enter the current source from node 3 to node 2. (Note that if you entered this source from node 2 to node 3, the sign is reversed: $-10 \pm 5^{\circ}A$.)

3 [ENTER] 2 I 10 [ENTER] 5 COMPLEX A

Enter the impedance from node 2 to node 3.

2	ENTE	R 3 Z	6 ENTER
4	+/-	COMPLEX	OHM

Location:	±	LENTED1	#

{ 0-Element Circuit }

ADD DEL PRINT INIT

Location:	#	[ENTER]	#

RCLZVI

Loca	tio	ז: #	[Eh	ITER]	#
	6		7	U	

Location: # [ENTER] #

C L Z V I

_ocation: # [ENTER] # RICILIZIY



Location: # [ENTER] # R C L Z V I

Location: # [ENTER] # RCLZVI



Press **EXIT** to return to the program's main menu.

At this point you can use \blacksquare and \bigtriangledown to view the elements in the circuit. Exit the "CIRCT" program and calculate the node voltages. If you want the results printed, press **PRINT A PON** to enable printing.

EXIT

XEQ NODA

Key in the number of nodes. (Do not include the reference node).

3 (R/S)

Since no capacitors or inductors were entered, the frequency is not needed; press [R/S] to calculate the results.

R/S

If you're not using a printer, press [R/S] to display the voltage at node 2.

R/S

If you're not using a printer, press [R/S] again to display the voltage at node 3.

R/S

x: 58.07 ∡126.80

No. Nodes? x: 0.00

Radian Frequency(2πf)? x:0.01

(6):2 - 3 6.00 - i4.00

EXIT

V2= × 45.46 ∡53.34

x 68.66 ∡47.03

Y: 1.00 X: 34.00

V1=

Impedance of a Ladder Network ("LADDR")

This program calculates the input impedance, Z_{in} , of a ladder network. Elements are added one at a time starting with the element furthest from the terminals where Z_{in} is measured. The first element must be connected in parallel.

Given an input *admittance* of Y_{in} , adding a shunt (parallel) R, L, or C results in a new input impedance:

$$Y_{new} = \begin{cases} Y_{in} + \left(\frac{1}{R_p} + j0\right) \\ Y_{in} + \left(0 - j\frac{1}{\omega L_p}\right) \\ Y_{in} + \left(0 + j\omega C_p\right) \end{cases}$$

Adding a series R, L, or C, we have:

$$Y_{new} = \begin{cases} \left(\frac{1}{Y_{in}} + (R_{o} + j0)\right)^{-1} \\ \left(\frac{1}{Y_{in}} + (0 + j\omega L_{o})\right)^{-1} \\ \left(\frac{1}{Y_{in}} + \left(0 - j\frac{1}{\omega C_{o}}\right)\right)^{-1} \end{cases}$$

where Y = 1/Z and $\omega = 2\pi f$.

Variable Used. Zin is the impedance looking into the ladder network.

Remarks.

 Elements are entered in rectangular form; however, the input impedance is displayed in polar form. (If you want the impedance displayed in rectangular mode, you can change line 29 of the "LADDR" program listing to 29 RECT and then delete line 31.)

- If a circuit element is given as a complex impedance, key in the complex value and then add it to the circuit using RP and RS.
- Flag 00 is set when $Z_{in} = 0$.
- Flag 21 (printer enable) is set or cleared to match flag 55 (printer existence). This automatically produces printer output if flag 55 is set.

Programs Required. "LADDR" (page 70), " $C \rightarrow Z$ " (page 148), and "FQ?" (page 149).

Program Instructions.

- 1. Key the required programs into your calculator.
- **2.** Press **XEQ** LADDR (to run the "LADDR" program).
- When you see Radian Frequency(2πf)?, key in the radian frequency and press R/S. (If the correct frequency is already in the X-register, just press R/S.)
- 4. The program displays a menu of the elements that can be added. For each element in the network (starting with the element furthest from the terminals where Z_{in} is measured), key in the value of the element and then press the corresponding key.

RP (parallel resistor, ohms)

LF (parallel inductor, henrys)

CP (parallel capacitor, farads)

RS (series resistor, ohms)

LS (series inductor, henrys)

CS (series capacitor, farads)

The series elements are not displayed until the first parallel element has been entered. Each time you add an element, the new impedance is displayed.

5. To quit, press EXIT. After quitting, you can restart "LADDR" by pressing R/S.

Example. Find the input impedance of the following circuit at a frequency of 1 MHz:



Select the FIX 2 display format and run the "LADDR" program.

DISP FIX 02 (XEQ) LADDR

Radian	Frequency(2πf)?
×:0.01	

The frequency is given in Hz. Convert it to radians/second.

E 6 🗐 🛪 2 🗙 R/S

Zin=	0.00) 40	.00	
RP	LP	CP		

Enter the four elements (working *right to left*). Notice that the current value of Z_{in} is displayed after each element is entered.

100 RP	Zin=100.00 ∡0.00 RP LP CP RS LS CS
650 E 12 +/_ CS	Zin=264.49 ∡-67.78 ₽ LP CP RS LS CS
120 E 6 +/_ LP	Zin=384.34 ∡-56.67 ℝP LP CP RS LS CS
1000 RP	Zin=306.73 ∡-41.82

RP LP CP RS LS

CS

The input impedance is $306.73 \pm -41.82^{\circ}$ ohms.

"LADDR" Program Listing.

Program:	Comments:	
00 (154-Byte Pr9m) 01 LBL "LADDR" 02 XEQ "FQ?"	Inputs the frequency using the "FQ?" utility on page 149.	
03 SF 00	Sets flag 00 (until the first element is entered).	
04 CLX 05 ENTER 06 COMPLEX 07 STO "Zin" 08 CF 21 09 FS? 55 10 SF 21	Initializes Z _{in} and sets or clears flag 21 to match flag 55.	
11 CLMENU 12 LBL A 13 "RP" 14 KEY 1 XEQ 01 15 "LP" 16 KEY 2 XEQ 02 17 "CP" 18 KEY 3 XEQ 03	Defines the menu keys for entering elements in parallel.	
19 FS? 00 20 GTO B 21 "RS" 22 KEY 4 XEQ 04 23 "LS" 24 KEY 5 XEQ 05 25 "CS"	If flag 00 is clear, declares the menu keys for entering elements in series.	
26 KEY 6 XEQ 06	Decodes the element location into two mesh or node numbers in the X- and Y-registers.	
27 28 29 30 31 32 33	LBL B MENU POLAR VIEW "Zin" RECT STOP GTO A	Displays the menu and the input impedance. (If you want the impedance displayed in rectangular mode, you can change line 29 to 29 RECT and delete line 31.)
--	---	---
34 35 36	LBL 01 1/X GTO 07	Converts a resistance to admittance.
37 38 39	LBL 02 XEQ "C→Z" GTO 07	Converts an inductance to admittance using the "C \rightarrow Z" utility on page 148.
40 41	LBL 03 XEQ "L→Z"	Converts a capacitance to admittance using the "L \rightarrow Z" utility on page 148.
42 43 45 46 47 48 49	LBL 07 RCL "Zin" FC?C 00 1/X + 1/X STO "Zin" RTN	Adds a parallel element.
50 51 52	LBL 06 XEQ "C→Z" GTO 04	Converts a capacitance to impedance.
53 54	LBL 05 XEQ "L→Z"	Converts an inductance to impedance.
55 56 57	LBL 04 STO+ "Zin" END	Adds a series element.

Filter Design

This chapter contains programs for calculating component values for standard active filters and for Butterworth filters between equal terminations.

Active Filter Design ("AF")

This program calculates element values for the standard active filter circuits shown below. You must provide F (the corner or center frequency), G (the midband gain), PF or α (the peaking factor), and C (a capacitor). The program then displays (and optionally prints) the list of elements that form the desired filter.

Low Pass Filter.



 $C_5 = C$

$$C_{2} = \frac{4C(G+1)}{PF^{2}}$$

$$R_{1} = \frac{PF}{4G\pi f_{0}C}$$

$$R_{3} = \frac{PF}{4\pi f_{0}C(G+1)} = \frac{G}{G+1}R_{1}$$

$$R_{4} = GR_{1}$$

High Pass Filter.



$$C_1 = C_3 = C$$

$$C_4 = \frac{C}{G}$$

$$R_2 = \frac{PF}{2\pi f_0 C \left(2 + \frac{1}{G}\right)}$$

$$R_5 = \frac{2G+1}{PF \, 2\pi f_0 C}$$

Bandpass Filter.



 $C_3 = C_4 = C$

$$R_1 = \frac{1}{G \, 2\pi f_0 C \, PF}$$

$$R_2 = \frac{1}{\left(\frac{2}{PF^2} - G\right) 2\pi f_0 C PF}$$

$$R_5 = \frac{2}{PF \ 2\pi f_0 C}$$

Variables Used.

In Equations	Description	In Program
f ₀	Center frequency (Hz).	F
G	Midband gain (db).	G
PF	Peaking factor.	PF
С	Capacitor (farads).	С

Remarks. Flag 21 (printer enable) is set by the program.

Program Instructions.

- 1. Key the "AF" program (listed on page 78) into your calculator.
- **2.** Press XEQ **AF** (to run the "AF" program).
- **3.** Press **INPUT**. The calculator prompts for F, G, PF, and C.
 - **a.** Key in a frequency, F, in Hertz; press **R/S**.
 - **b.** Key in the midband gain, G; press **R/S**.
 - **c.** Key in a peaking factor, PF; press \mathbb{R}/\mathbb{S} .
 - **d.** Key in a capacitance, C, in Farads; press **\mathbb{R}/S**.

After entering these values, the program returns to the main menu.

- 4. Press LOWP (low pass), HIGHP (high pass), or EAND (band pass) to calculate the elements needed to build the particular filter.
- 5. When all of the elements have been displayed, press **R/S** to return to the main menu. Then go to step 3 to work another problem or press **EXIT** to quit.

Example. Design a high-pass active filter with the following characteristics: F = 10 Hz, G = 10, PF = 1, and $C = 1 \mu$ F.

Select the ENG 3 display format and run the "AF" program.

DISP ENG 03 (XEQ) AF

x: 0.000E0	
LOWP HIGHP BAND	INPUT

INPUT	Y: 0.000E0 F?0.000E0
10 (<u>R/S</u>)	Y: 10.0050 G?0.00050
10 (<u>R/S</u>)	Y: 10.00E0 PF?0.000E0
1 <u>R/S</u>	Y: 1.000E0 C?0.000E0
E 6 +/-	X: 1.000E-6
	LOWP HIGHP BAND
HIGHP	Сомер (шане (зимо R2= x: 7.579е3
HIGHP If you're not using a printer, press R/S af	R2= x: 7.579E3 ter each result.
HIGHP If you're not using a printer, press R/S af R/S	R2= x: 7.579E3 ter each result. R5= x: 334.2E3
HIGHP If you're not using a printer, press R/S af R/S R/S	R2= x: 7.579E3 ter each result. R5= x: 334.2E3 C1=C3=C= x: 1.000E-6

[R/S]	X: 100.0E-9
	LOWP HIGHP EAND

"AF" Program Listing.

Program:

00 (309-Byte Prom) 01 LBL "AF" 02 SF 21 03 CLMENU 04 "LOWP" 05 KEY 1 XEQ A 06 "HIGHP" 07 KEY 2 XEQ B 08 "BAND" 09 KEY 3 XEQ C 10 "INPUT" 11 KEY 6 XEQ I 12 KEY 9 GTO 09 13 LBL 00 14 CLD 15 MENU 16 STOP 17 GTO 00 18 LBL A 19 "R1" 20 RCL "PF" 21 4 22 RCLX "G" 23 XEQ 04 24 ÷ 25 XEQ 08 26 "R3" 27 RCL× "G" 28 ENTER 29 ENTER 30 LASTX 31 1 32 + 33 ÷ 34 XEQ 08

Comments:

Defines the menu for selecting a filter type.

Clears any message that may be displayed, displays the menu and stops. Pressing [R/S] redisplays the menu.

Calculates the elements for a low pass filter.

35 "R4" 36 R+ 37 XEQ 08 38 "C2" 39 1 40 RCL "G" 41 RCL× "C" 42.4 43 X 44 RCL "PF" 45 X+2 46 ÷ 47 XEQ 08 48 "C5=C" 49 LBL 10 50 RCL "C" 51 GTO 08 52 LBL B 53 "R2" 54 2 55 XEQ 04 56 RCL "G" 57 1/X 58 2 59 + 60 X 61 RCL÷ "PF" 62 17X 63 XEQ 08 64 "R5" 65 2 66 RCL× "G" 67 1 68 + 69 2 70 XEQ 04 71 RCL× "PF" 72 ÷ 73 XEQ 08 74 "C1=C3=C"

Calculates the elements for a high pass filter.

75	XEQ 10
76	"C4"
77	RCL÷ "G"
78	GTO 08
79	LBL C
80	"R1"
81	2
82	XEQ 04
83	RCL× "G"
84	ENTER
85	ENTER
86	RCL× "G"
87	178
88	XEQ 08
89	"R2"
90	CLX
91	2
92	RCL "PF"
93	X+2
94	÷
95	RCL- "G"
96	×
97	17X
98	XEQ 08
99	"R5"
100) CLX
101	. 2
102	2 X<>Y
103	; ÷
104	H XEQ 08
105) "C3=C4=C"
106	610-10
107	'LBL 04
108	3 PI
109	9 RCL× "F"
110) RCL× "C"
111	×
112	RTN

Calculates the elements for a band pass filter.

Calculates πFC .

113 LBL I 114 EXITALL 115 INPUT "F" 116 INPUT "G" 117 INPUT "PF" 118 INPUT "C" 119 RTN 120 LBL 08 121 EXITALL 122 F"=" 123 FS? 55 124 ARCL ST X 125 AVIEW 126 LBL 09 127 END Exits all menus and inputs F, G, PF, and C.

Exits all menus. If results are being printed (flag 55 set), puts label and value on the same line.

Ends program.

Butterworth Filter Design ("BF")

This program calculates component values for Butterworth filters between *equal* terminations. Inputs are termination resistance, passband characteristics, and attenuation at some out-of-band frequency.

Before the filter elements can be calculated, a normalized frequency must be computed from the desired cutoff or center frequency and passband characteristics. The normalized frequency is computed by one of these formulas:

Low Pass

High Pass

 $\omega_n = \frac{\omega}{\omega_0} \qquad \qquad \omega_n = \frac{\omega_0}{\omega}$

Band Pass

Band Elimination

 $\omega_n = \frac{\omega^2 - \omega_0^2}{BW\omega} \qquad \omega_n = \frac{BW\omega}{\omega_0^2 - \omega^2}$

The basic form of the filter is this low-pass prototype:



whose elements are given by the following set of formulas:

$$C_{i} = \frac{1}{\pi f_{c} R} \sin \frac{(2i - 1)\pi}{2n}, i = 1, 3, 5, ..., n - 1$$
$$L_{i} = \frac{R}{\pi f_{c}} \sin \frac{(2i - 1)\pi}{2n}, i = 2, 4, 6, ..., n$$

where:

$$n = INT \left[\frac{1 + \ln(2 \times 10^{-\Delta dB/10} - 1)}{2\ln(\omega/\omega_0)} \right]$$

Once the low-pass values have been calculated, if another passband characteristic is desired, the filter components are changed by one of the frequency transformations shown on the next page.



84 3: Butterworth Filter Design

Variables and Storage Registers Used.

In Equations	Description	In Program
R	Resistor (ohms).	R
F ₀	Center frequency (Hz).	F0
ω_0	Center frequency (radians/sec.).	R ₀₁
F ₁	Attenuation frequency (Hz).	F1
	Amount of attenuation (dB).	A
BW	Band width (Hz).	вм
	Band width (radians/sec.).	R ₀₂
	Filter type (1-4).	R ₀₇
n	Filter order.	R ₀₉
i	Element counter.	R ₁₁

Remarks.

- Flag 01 (set and cleared by the program) is used for branch control.
- Flag 21 (*printer enable*) is set by the program.
- Registers R₀₈, R₁₀, R₁₃, and R₁₄ are used to store intermediate results. Be sure to set the SIZE to at least 15 registers (MODES)
 SIZE 15 ENTER).
- The "BF" program sets Radians mode.



The program will give erroneous results if asked to calculate a filter order when A is small (when ΔdB is close to Loss (ω_0)).

Program Instructions.

- 1. Key the "BF" program (listed on page 88) into your calculator.
- **2.** Press **XEQ BF** (to run the "BF" the program).
- **3.** The program displays a variable menu containing *R*, *F0*, *F1*, *A*, and *BW*. Store a value into each variable by keying in the value and then pressing the corresponding menu key.
- **4.** After each of the five values has been stored, press \mathbb{R}/\mathbb{S} .
- 5. The program displays Type? and a menu containing the four types of filters. Press one of these keys to select a filter type:

LOWP (low pass)

HIGHP (high pass)

BPASS (band pass)

BELIM (band elimination)

6. The program then calculates and displays N and the filter elements. If you're not using a printer, press **R/S** after each result is displayed.

Example. Design a 100-Hz wide Butterworth filter centered at 800 Hz with a 30-dB attenuation at 900 Hz. R_0 is 50 ohms. The termination resistance, R, is also 50 ohms.

Select ENG 3 display format and run the "BF" program.

DISP ENG 03 (XEQ) BF	X: 0.000E0 R F0 F1 A EW
Store the five inputs.	
50 R	R=50.00E0 R F0 F1 A BH
800 FØ	F0=800.0E0
900 F 1	F1=900.0E0

30 A

A=30.00E0 R F0 F1 A EW

100 BW

BW=1	00.1	0e0			
6	FO	F1	Ĥ	BM	

Continue with the program.

R/S

Type? Lowp (Highp (Spass (Selim)

Select a band-pass filter. If you are not using a printer, press [R/S] after each result is displayed.

BPASS	N= x: 6.000E0
(R/S)	C1=16.48E-6 L1=2.402E-3
(R/S)	L2=112.5E-3 C2=351.7E-9
[R/S]	C3=61.49E-6 L3=643.6E-6
(R/S)	L4=153.7E-3 C4=257.5E-9

R/S	C5=45.02E-6 L5=879.2E-6

Therefore, the filter you've calculated looks like this:



"BF" Program Listing. This program demonstrates that local labels do not have to be unique within a program as long as careful consideration is given to the local label search order (described in the owner's manual).

Program:

Comments:

01	LBL "BF"	Declares the menu variables and sets
02	MVAR "R"	Radians mode.
Ø3	MVAR "F0"	
04	MVAR "F1"	
05	MVAR "A"	
Ø6	MVAR "BW"	
07	RAD	
08 09	LBL A VARMENU "BF"	Displays the variable menu and stops. Pressing R/S is the only way
08 09 10	LBL A VARMENU "BF" CLA	Displays the variable menu and stops. Pressing $\boxed{R/S}$ is the only way to continue the program.
08 09 10 11	LBL A VARMENU "BF" CLA STOP	Displays the variable menu and stops. Pressing $[\overline{R/S}]$ is the only way to continue the program.
08 09 10 11 12	LBL A VARMENU "BF" CLA STOP ALENG	Displays the variable menu and stops. Pressing R/S is the only way to continue the program.
08 09 10 11 12 13	LBL A VARMENU "BF" CLA STOP ALENG X≠0?	Displays the variable menu and stops. Pressing R/S is the only way to continue the program.

```
15 RCL "FØ"
                                 Converts the center frequency to
16 XEQ 10
                                 radians/second.
17 STO 01
18 CLMENU
                                 Defines the menu for selecting a
19 "LOWP"
                                 filter type. The EXIT key is defined
20 KEY 1 GTO C
                                 to return to the variable menu
21 "HIGHP"
22 KEY 2 GTO D
23 "BPASS"
24 KEY 3 GTO E
25 "BELIM"
26 KEY 4 GTO F
27 KEY 9 GTO A
28 LBL B
                                 Displays the menu and prompts for a
29 MENU
                                 filter type. Pressing R/S redisplays
30 "Type?"
                                 the menu.
31 PROMPT
32 GTO B
33 LBL C
                                 Calculates the elements of a low pass
                                 filter (type 1).
34-1
35 GTO 01
36 LBL D
                                 Calculates the elements of a high
                                 pass filter (type 2).
37 2
38 GTO 01
39 LBL E
                                 Calculates the elements of a band
                                 pass filter (type 3).
40 3
41 GTO 01
42 LBL F
                                 Calculates the elements of a band
43 4
                                 elimination filter (type 4).
44 LBL 01
                                 Exits all menus, stores the filter type,
                                 and sets flag 21 for proper output.
45 EXITALL
46 STO 07
47 SF 21
```

```
48 RCL "BW"
49 XEQ 10
50 STO 02
51 LBL 00
52 RCL "F1"
53 RCL "A"
54 10
55 ÷
56 10+X
57 2
58 X
59 1
60 -
61 LN
62 STO 08
63 X<>Y
64 XEQ 10
65 XEQ 07
66 RCL 08
67 X<>Y
68 LN
69 ABS
70 ÷
71 1
72 +
73 2
74 ÷
75 IP
76 STO 09
77 STO 10
78 "N="
79 FS? 55
80 AIP
81 AVIEW
82 ADV
83 LBL 08
84 RCL 09
85 RCL- 10
```

Converts the band width to radians/second.

Calculates the filter order.

Displays the filter order. If results are being printed (flag 55 set), the label and result are displayed on the same line.

Evaluates the Butterworth equations.

Calculates the frequency transformation for the particular filter type.

126 17X 127 RCL 01 128 X+2 129 ÷ 130 XEQ 06 131 +/-132 GTO 00 133 LBL 04 134 SF 01 135 RCL× 02 136 RCL 01 137 X+2 138 ÷ 139 XEQ 06 140 XEQ 00 141 ABS 142 RCL 01 143 X+2 144 × 145 1/X 146 XEQ 06 147 +/-148 LBL 00 149 FS? 01 150 CLA 151 Xè0? 152 F"L" 153 XK0? 154 ⊢"C" 155 RCL 11 156 AIP 157 ⊢"=" 158 R+ 159 ABS 160 ARCL ST X 161 - "4" 162 FC? 01 163 AVIEW 164 FS?C 01 165 RTN

Displays a pair of filter elements.

166	DSE	10	
167	GTO	Ø8	
168	RTN		
169	LBL	07	
170	STO	14	
171	GTO	IND	07
172	LBL	04	
173	XEQ	03	
174	GTO	00	
175	LBL	Ø2	
176	XEQ	01	
177	LBL	00	
178	1/X		
179	+/-		
180	GTO	05	
181	LBL	01	
182	RCL	14	
183	RCL+	÷ 01	
184	GTO	05	
185	LBL	03	
186	RCL	14	
187	X+2		
188	RCL	01	
189	X+2		
190	-		
191	RCL-	÷ 14	
192	RCL-	÷ 02	
193	LBL	05	
194	ABS		
195	STO	13	
196	RTN		
197	LBL	06	
198	-1		
199	RCL	11	
200	Y≁X		
201	×		
202	RTN		

Repeats the Butterworth calculations for each element.

Calculates the normalized frequency for the particular filter type.

Multiplies by -1^{i} .

Converts the frequency in the X-register (in Hz) to radians/second.

Transmission Lines

This chapter contains programs that calculate the impedance of a lossy high-frequency transmission line and the high-frequency characteristic impedances for five types of transmission line configurations.

Transmission Line Calculations ("LINE")

This program calculates the input impedance of a lossy transmission line terminated in Z_L . The program provides an exact solution when the distributed line parameters R_0 (defined as $\sqrt{L \div C}$), R, and G are given. It provides an approximate solution when R_0 and the conductor and dielectric losses are given.



The transmission line shown has a lumped model composed of elements L, C, R, and G as follows:



From this model the following equations can be derived:

$$R_{0} = \sqrt{\frac{L}{C}}$$
$$r = \frac{R}{L} = \frac{vR}{R_{0}}$$
$$g = \frac{G}{C} = vR_{0}G$$

where:

- L = inductance/unit length.
- C = capacitance/unit length.
- G = conductance/unit length.
- R = resistance/unit length.

$$v = 3 \ge 10^8 v_r$$
.

- v_r = relative phase velocity.
- f = frequency, Hz.
- $\omega = 2\pi f$ radians/second.

and

$$\alpha = \frac{1}{\sqrt{2v}} \left[rg - \omega^2 + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \right]^{1/2}$$
$$\beta = \frac{1}{\sqrt{2v}} \left[\omega^2 - rg + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \right]^{1/2}$$

The approximate solution is:

$$\operatorname{Re} \left\{ Z_{0} \right\} = R_{0} \left[1 + \frac{1}{2} \left(\frac{\alpha_{C} - \alpha_{D}}{\beta_{0}} \right) \left(\frac{3 \alpha_{D} + \alpha_{C}}{\beta_{0}} \right) \right]$$
$$\operatorname{Im} \left\{ Z_{0} \right\} = R_{0} \left[\frac{\alpha_{D} - \alpha_{C}}{\beta_{0}} \right]$$
$$\alpha = \alpha_{C} + \alpha_{D}$$

$$\beta = \beta_0 \left[1 + \frac{1}{2} \left(\frac{\alpha_C - \alpha_D}{\beta_0} \right)^2 \right]$$

where:

 α_C = conductor loss, nepers/unit length = $0.5(R/R_0)$

 α_D = dielectric loss, nepers/unit length = 0.5(GR)

 $\beta_0 = \omega/v$

Then

$$Z_{in} = Z_0 \left(\frac{1 + \Gamma_L e^{-2\gamma l}}{1 - \Gamma_L e^{-2\gamma l}} \right) \text{ohms.}$$

where:

$$\Gamma_L = \frac{Z_L - Z_0}{Z_L + Z_0}$$

l = line length

 Z_L = impedance of termination (ohms)

 Z_0 = characteristic impedance of line = Re{ Z_0 } + j Im{ Z_0 } (ohms)

 γ = propagation constant of line = α + $j\beta$

98 4: Transmission Line Calculations

 Z_0 and γ are computed differently depending on which solution is selected.

$$\operatorname{Re} \{ Z_0 \} = \frac{R_0}{\sqrt{2(g^2 + \omega^2)}} \Big[rg + \omega^2 + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \Big]^{1/2}$$
$$\operatorname{Im} \{ Z_0 \} = \frac{\pm R_0}{\sqrt{2(g^2 + \omega^2)}} \Big[-(rg + \omega^2) + \sqrt{(r^2 + \omega^2)(g^2 + \omega^2)} \Big]^{1/2}$$

The + sign is chosen when $g \ge r$ and the - sign is chosen when g < r.

Variables Used.

In Equations	Description	In Program
f	Frequency.	f
v _r	Relative phase velocity.	۷r
R ₀	Characteristic impedance.	RØ
1	Line length.	1
Z_L	Impedance of termination.	ZL

Remarks.

- Flag 21 (printer enable) is set or cleared to match flag 55 (printer existence). This automatically produces printer output if flag 55 is set.
- Registers R₀₀ thru R₀₆ are used for storing intermediate results. Be sure to set the SIZE to at least seven registers (MODES) V
 SIZE 7 ENTER) before running "LINE".

Program Instructions.

- 1. Key the "LINE" program (listed on page 102) into your calculator.
- **2.** Select the desired angular and coordinate modes.
- **3.** Press **XEQ** LINE (to run the "LINE" program). The program displays a variable menu containing f, ν_r , R_0 , l, and Z_L (displayed as F, VR, R0, L, and ZL).
- **4.** Store a value into each variable by keying in the value and pressing the corresponding menu key.

- **5.** After all five variables have been stored, press **R/S**.
- 6. Select an exact or approximate solution.

For an exact solution:

- a. Press EXACT.
- **b.** The program prompts for G. Key in the conductance value and press $\overline{\mathbb{R}/\mathbb{S}}$.
- **c.** The program then prompts for R. Key in the resistance value and press [R/S].

For an approximate solution:

- a. Press APROX.
- **b.** The program prompts for the conductor loss. Key in the value and press **R/S**.
- **c.** The program then prompts for the dielectric loss. Key in the value and press **R/S**.
- 7. The value for Z_{in} is calculated and displayed. To calculate another solution, go to step 6. To work a new problem, press **EXIT** and go to step 4.

Example. A transmission line has the following properties:

R = 1.2664 ohms/cm.

G = 0.00004187 siemens/cm.

 $R_0 = 55$ ohms.

 $v_r = 0.85.$

What is the input impedance of 3.5 cm of this line at 2 GHz if it is terminated in $Z_L = 75 \not a - 30^\circ$ ohms?

Select Degrees and Polar modes, select FIX 2 display format, and run the "LINE" program.

MODES	DEG	MODES	POLAR	×:0.00				
DISP	FIX 02	XEQ LI	NE	F VR	RŬ	L	ZL	

2 E 9	F
-------	---

f=2,	000	,000	,000	.00	
F	VR	RO	L	ZL	

.85 ¥R	Vr=0.85 F VR RO L ZL
55 RØ	R0=55.00 F VR R0 L ZL
3.5 L	1=3.50 F VR RO L ZL
75 (ENTER) 30 (*/-) COMPLEX ZL	ZL=75.00 ∡-30.00 F VR R0 L 2L
(R/S)	x: 2.75 BXM01
EXACT	Y: 2.75 G?0.00
4.187 E 5 +/_ R/S	Y: 4.19E-5 R?0.00
1.2664 (R/S)	Zin=48.01 ∡28.48

The required input impedance is 48.01 \angle 28.48° ohms.

"LINE" Program Listing.

Program:

00 { 404-Byte Prom } 01 LBL "LINE" 02 MVAR "f" 03 MVAR "Vr" 04 MVAR "R0" 05 MVAR "1" 06 MVAR "ZL" 07 CF 21 08 FS? 55 09 SF 21 10 LBL A 11 CLA 12 VARMENU "LINE" 13 STOP 14 ALENG 15 X≠0? 16 GTO A 17 RCL "f" 18 1€10 19 ÷ 20 STO 04 21 2 22 PI 23 X 24 X 25 STO 03 26 RCL "1" 27 2 28 X 29 3 30 RCLX "Vr" 31 STO 00 32 ÷ 33 STO 02

Comments:

Defines the menu variables and sets or clears flag 21 to match flag 55.

Displays the variable menu and stops. The program continues only when $[\overline{R/S}]$ is pressed.

Calculates intermediate results used by both solutions.

34	LBL B	
35	CLMENU	
36	"EXACT"	
37	KEY 1 XEQ	С
38	"APROX"	
39	KEY 6 XEQ	D
40	KEY 9 GTO	Ĥ
41	MENU	
42	STOP	
43	GTO B	
44	LBL C	
45	EXITALL	
46	INPUT "G"	
47	INPUT "R"	
48	RCL× 00	
49	RCL÷ "RØ"	
50	STO 01	
51	RCL "G"	
52	RCL× "R0"	
53	STOX 00	
54	RCL 03	
55	RCL 01	
56	→POL	
57	SQRT	
58	STO 01	
59	X<>Y	
60	2	
61	÷	
62	STO 06	
63	RCL 03	
64	RCL 00	
65	→POL	
66	SQRT	
67	STO 00	
68	X<>Y	
69	2	
70	÷	
71	STO 03	
72	RCL+ 06	

Displays a menu for selecting the type of solution. The **EXIT** key is defined to go back to the variable menu.

Calculates the "exact" solution.

```
73 STO 05
74 RCL 06
75 RCL- 03
76 STO "Vr"
77 RCL 01
78 RCL÷ 00
79 STO× "RØ"
80 RCL 01
81 RCL× 00
82 STOX 02
83 GTO E
84 LBL D
85 EXITALL
86 CLMENU
87 "C loss?"
88 PROMPT
89 STO 00
90 "D loss?"
91 PROMPT
92 STO 03
93 RCL "1"
94 RCL "Vr"
95 RCL 04
96 PI
97 ×
98 1.5
99 ÷
100 X<>Y
101 ÷
102 STO 06
103 ×
104 2
105 ×
106 STO 02
107 RCL 03
108 10
109 LN
110 20
111 ÷
```

Calculates the "approximate" solution.

146 LBL E 147 POLAR 148 RCL "ZL" 149 COMPLEX 150 STO 06 151 X<>Y 152 STO 01 153 RCL 05 154 RCL 02 155 →REC 156 +/-157 E+X 158 STO 02 159 X<>Y 160 180 161 × 162 PI 163 ÷ 164 STO 03 165 RCL 06 166 RCL- "Vr" 167 RCL 01 168 RCL÷ "RØ" 169 →REC 170 1 171 + 172 →POL 173 1/X 174 - 2175 × 176 X<>Y 177 +/-178 X<>Y 179 →REC 180 1 181 + 182 →POL 183 RCL× 02

Completes the calculations for both solutions.
```
184 17X
185 X<>Y
186 RCL- 03
187 +/-
188 X<>Y
189 →REC
190 1
191 -
192 →POL
193 1/X
194 2
195 ×
196 X<>Y1
197 +/-
198 X<>Y
199 →REC
200 1
201 +
202 →POL
203 RCL× "R0"
204 X<>Y
205 RCL "Vr"
206 +
207 COMPLEX
208 "Zin="
209 ARCL ST X
210 AVIEW
211 END
```

Displays the result and returns to the menu at LBL B.

Transmission Line Impedance ("TLI")

This program calculates the high frequency characteristic impedance (Z_0) for five types of transmission lines:

Open two-wire line:
$$Z_0 = \frac{120}{\sqrt{\epsilon_r}} \ln\left(\frac{2D}{d}\right)$$
Single wire near ground: $Z_0 = \frac{138}{\sqrt{\epsilon_r}} \log\left(\frac{4h}{d}\right)$ Balanced wires near
ground: $Z_0 = \frac{276}{\sqrt{\epsilon_r}} \log\left\{\frac{2D}{d}\left[1 + \left(\frac{D}{2h}\right)^2\right]^{-1/2}\right\}$ Wires in parallel near
ground: $Z_0 = \frac{69}{\sqrt{\epsilon_r}} \log\left\{\frac{4h}{d}\left[1 + \left(\frac{2h}{D}\right)^2\right]^{+1/2}\right\}$ Coaxial line: $Z_0 = \frac{60}{\sqrt{\epsilon_r}} \ln \frac{D}{d}$ Variables Used.

In Equations	Description	In Program
D	Wire spacing.	D
d	Wire diameter.	dia
h	Height of wire (above ground).	h
ϵ_r	Relative permittivity.	е

Remarks. Flag 21 (*printer enable*) is set or cleared to match flag 55 (*printer existence*). This automatically produces printer output if flag 55 is set.

Program Instructions.

- 1. Key the "TLI" program (listed on page 111) into your calculator.
- **2.** Press XEQ TLI (to run the "TLI" program).
- **3.** Use the variable menu displayed by the program to store the required inputs for the particular line configuration you're working with.
- 4. Press R/S.
- 5. The program displays Line Config? and a menu containing the five types of line configurations. Select a configuration by pressing the corresponding menu key:

OP (open two-wire line)

SW (single wire near ground)

B (balanced wires near ground)

F (wires in parallel near ground)

C (coaxial line)

6. The value of Z_0 is displayed and the program returns to the input menu (step 3).

Example. Calculate Z_0 of RG-218/U coaxial cable with wire spacing, D = 0.68 in., wire dia = 0.195 in., and ε_r of the polyethylene insulation = 2.3.

Select FIX 2 display format and run the "TLI" program.

DISP FIX 02 (XEQ) TLI

x: 0.	00			
D	DIA	H	Ε	

Store the inputs.

.68 D

D=0.	68			
D	DIA	H	Ε	

.195 DIA

2.3 E

Calculate Z_0 for the *coaxial* configuration. [R/S]

106	1 5	4	в		C .	

Line Config?

D=6.00

dia=0.08

Now calculate Z_0 for an open two-wire (air) line with D = 6 in. and dia = 0.0808 in. (ε_r of air = 1).

6 D

.0808 DIA

C

D DIA H E

D DIA H E

е=1.00 о ой н е

OP SW B P C

ine Config?

R/S

0P 20=600.08

е=2.30 D DIM H E

dia=0.20 D DA H E

20=49.42 0 000 H E

1 E

This time, calculate Z_0 of an air line (ϵ_r of air = 1) consisting of a single 0.1285 inch diameter wire 6 inches from a ground plane.

.1285 DIA	dia=0.13 0 OM H E
6 H	h=6.00 о он н с
1 E	e=1.00 0 0M H E
R/S)	Line Config? OP SW 8 P C
SW	20=313.44 о оп н е

"TLI" Program Listing.

Program:

00	(225-Byte	Prgm	Э
01	LBL "TLI"		
02	MVAR "D"		
03	MVAR "dia"		
04	MVAR "h"		
05	MVAR "e"		
Ø6	CF 21		
07	FS? 55		
08	SF 21		

Comments:

Declares menu variables and sets or clears flag 21 to match flag 55.

09 LBL A 10 VARMENU "TLI" 11 CLA 12 STOP 13 ALENG 14 X≠0? 15 GTO A 16 CLMENU 17 "OP" 18 KEY 1 XEQ 01 19 "SW" 20 KEY 2 XEQ 02 21 "B" 22 KEY 3 XEQ 03 23 "P" 24 KEY 4 XEQ 04 25 "C" 26 KEY 5 XEQ 05 27 KEY 9 GTO A 28 MENU 29 "Line Config?" 30 PROMPT 31 GTO A 32 LBL 01 33 XEQ 07 34 LN 35 120 36 GTO 06 37 LBL 02 38 XEQ 08 39 LOG 40 138 41 GTO 06

Displays the variable menu and stops.

Defines the programmable menu for the five configurations. The **EXIT** key is defined to return to the variable menu.

Displays the menu and prompts for a selection. After the appropriate routine is executed, returns to the variable menu.

Calculates Z_0 for an open two-wire configuration.

Calculates Z_0 for a single wire near the ground.

42	LBL	03
43	XEQ	09
44	1/X	
45	1	
46	+	
47	SQRI	Г
48	1/X	
49	XEQ	07
50	×	
51	LOG	
52	276	
53	GTO	06
54	LBL	04
55	XEQ	09
56	1	
57	+	
58	XEQ	08
59	SQR	Г
60	XEQ	08
61	×	
62	LOG	
63	69	
64	GTO	06
65	LBL	05
66	RCL	"D"
67	RCL	÷ "dia"
68	LN	
69	60	
70	LBL	06
71	RCL	"e"
72	SQR	Т
73	÷	
74	×	
75	"ZØ:	="
76	ARCI	L ST X
77	AVI	EW
78	RTN	

Calculates Z_0 for balanced wires near the ground.

Calculates Z_0 for wires in parallel near the ground.

Calculates Z_0 for a coaxial line configuration.

Completes the calculation and displays the result.

79 80 81 82 83	LBL 07 2 RCL× "D" RCL÷ "dia" RTN	Calculates an intermediate result.
84 85 86 87 88	LBL 08 4 RCL× "h" RCL÷ "dia" RTN	Calculates an intermediate result.
89 90 91 92 93 94	LBL 09 2 RCL× "h" RCL÷ "D" X≁2 END	Calculates an intermediate result.

Amplifier Analysis

This chapter contains programs that calculate small-signal properties of a transistor amplifier and automate a method of transistor bias optimization.

Transistor Amplifier Performance ("TAP")

This program calculates certain small-signal properties of a transistor amplifier given the h-parameter matrix and the source and load impedances. The program calculates current and voltage gains, and input and output impedances.



Equations. The definition of the *h*-parameter matrix is:

$$\begin{bmatrix} V_1 \\ I_2 \end{bmatrix} = \begin{bmatrix} h_i & h_r \\ h_f & h_o \end{bmatrix} \begin{bmatrix} I_1 \\ V_2 \end{bmatrix}$$

The current gain is:

$$A_{i} = \frac{i_{2}}{i_{1}} = \frac{-h_{f}}{1 + h_{o} Z_{L}}$$

The voltage gain is:

$$A_v = \frac{v_2}{v_1} = \frac{A_i Z_L}{Z_{in}}$$

The voltage gain with a source resistor is:

$$A_{vo} = \frac{v_2}{v_S} = \frac{A_i Z_L}{Z_{in} + Z_S}$$

116 5: Transistor Amplifier Performance

The input impedance is:

$$Z_{in} = h_i + h_r Z_L A_i$$

The output impedance is:

$$Z_{out} = \frac{h_i + Z_S}{h_o h_i + h_o Z_S - h_f h_r}$$

Variables Used.

In Equations	Description	In Program
h _i	Matrix h-parameter.	hi
h _r	Matrix h-parameter.	hr
hf	Matrix h-parameter.	hf
h _o	Matrix h-parameter.	ho
Z_S	Source impedance.	Zs
	Load impedance.	Z1

Remarks. Flag 21 (*printer enable*) is set or cleared to match flag 55 (*printer existence*). This automatically produces printer output if flag 55 is set.

Program Instructions.

- 1. Key the "TAP" program (listed on page 120) into your calculator.
- **2.** Press **XEQ THF** (to run the "TAP" program).
- **3.** Use the variable menu displayed by the program to store the four h-parameter values $(h_i, h_r, h_f, \text{ and } h_o)$ and the source and load impedances $(Z_S \text{ and } Z_L)$.
- 4. Press **R/S** to display the result menu.
- **5.** Press the appropriate menu key for each result you want to calculate:
 - FI to display the current gain, A_i .
 - AV to display the voltage gain, A_v .

- **AVS** to display the voltage gain with a source resistor, A_{ve} .
- **ZIN** to display the input impedance, Z_{in} .
- ZOUT to display the output impedance, Z_{out} .
- 6. Press **EXIT** to return to the variable menu (step 3).

Example. What are the small-signal properties of a transistor that has the following h-parameter matrix and has source and load impedances of 1,000 and 10,000 ohms, respectively?

$$h = \begin{bmatrix} 1100 \ 250E - 6 \\ 50 \ 25E - 6 \end{bmatrix}$$

Select the FIX 4 display format and run the "TAP" program.

DISP FIX 04 XEQ TAP

X: 0.0000 HD HR HF HD ZS ZL

Using the variable menu, enter each of the six input values and then press [R/S].

1100 HI	hi=	1,10	0.00	00		
	HI	HR	HF	HO	ZS	ZL

250 E 6 +/- HR

50 HF

hf=5	50.00	000			
HI	HR	HF	HO	25	ZL

HI HR HF HO ZS ZL

hr=0.0003

25 E 6 +/- HO

ho=2.5000E-5 HI HR HF HD ZS ZL

1000 = -	7 4 000 0000
1000 23	25=1,000.0000
	HI HR HF HD ZS ZL

X: 52,500.0000 HI HR HF HD 25 ZL

10000 ZL

R/S

For each of the outputs you want to calculate, press the corresponding menu key.

AI	Ai=-40.0000 Ai AV AVS ZIN ZOUT
AV	Av=-400.0000 Al av avs zin zout
AVS	Avs=-200.0000 Al AV AVS ZIN ZOUT
ZIN	Zin=1,000.0000 Al AV AVS ZIN ZOUT
ZOUT	Zout=52,500.0000 AI AV AVS ZIN ZOUT
Press EXIT to return to the input menu.	

EXIT)

From here you can work another problem or press **EXIT** again to quit.

Z1=1	0,00	00.0	000		
HI	HR	HF	HO	ZS	ZL

X: 0.0000 Al AV AVS ZIN ZOUT

"TAP" Program Listing.

Program:

00 (227-Byte Prom) 01 LBL "TAP" 02 MVAR "hi" 03 MVAR "hr" 04 MVAR "hf" 05 MVAR "ho" 06 MVAR "Zs" 07 MVAR "Z1" 08 CF 21 09 FS? 55 10 SF 21 11 LBL A 12 VARMENU "TAP" 13 CLA 14 STOP 15 ALENG 16 X≠0? 17 GTO A 18 CLMENU 19 "Ai" 20 KEY 1 XEQ 01 21 "Av" 22 KEY 2 XEQ 02 23 "Aus" 24 KEY 3 XEQ 03 25 "Zin" 26 KEY 4 XEQ 04 27 "Zout" 28 KEY 5 XEQ 05 29 KEY 9 GTO A 30 LBL 00 31 MENU 32 STOP 33 GTO 00

Comments:

Declares the menu variables and sets or clears flag 21 to match flag 55.

Displays the variable menu and stops.

Defines the programmable menu for displaying the results. The **EXIT** key is defined to return to the variable menu.

Displays the programmable menu and stops. The menu is redisplayed after each result.

34 35 36 37	LBL 01 XEQ 07 "Ai" GTO 06	Calculates A _i .
38 39 40 41 42	LBL 02 XEQ 08 ÷ "Av" GTO 06	Calculates A _v .
43 44 45 46 47 48	LBL 03 XEQ 08 RCL+ "Zs" ÷ "Avs" GTO 06	Calculates A _{ve} .
49 50 51 52	LBL 04 XEQ 08 "Zin" GTO 06	Calculates Z _{in} .
53 54 55 57 58 60 61 63 64 65 66	LBL 05 RCL "hi" RCL+ "Zs" LASTX RCL "ho" × LASTX RCL× "Zs" + RCL "hf" RCL× "hr" - ÷	Calculates Z _{out} .

Displays a result. 67 LBL 06 68 ⊢"=" 69 ARCL ST X 70 AVIEW 71 RTN 72 LBL 07 Calculates A_i . 73 RCL "hf" 74 +/-75 1 76 RCL "ho" 77 RCL× "Z1" 78 + 79 ÷ 80 RTN Calculates Z_{in} . 81 LBL 08 82 XEQ 07 83 RCL× "Z1" 84 ENTER 85 RCL× "hr" 86 RCL+ "hi" 87 END

Transistor Amplifier Bias Optimization ("BIAS")

This program automates the method of bias optimization described in "Designing Class 'A' Amplifiers to Meet Specified Tolerances," by Ward J. Helms (*Electronics*, August 8, 1974). The program requires you to specify a set of parameters from which it determines, by an iterative technique, the optimum values for R_1, R_2, R_E , and R_L . The minimum power gain is also computed.



How "BIAS" Works. First, values are input for the variables listed in the table on page 126. Then, the transistor's thermal resistance is calculated:

$$\theta_{JA} = (T_{\text{max}} - 25^{\circ}C) / P_D$$

and the minimum load resistance and emitter resistance are estimated:

$$R_{L1} = \frac{\theta_{JA} V_{CC}^2}{4.4 (T_{J \max} - T_{A \max})} = R_{Ln}$$
$$R_{E1} = 0.1 R_{L1} = R_{En}$$

Next, the quiescent, maximum, and minimum collector currents are calculated:

$$I_{CQ} = \frac{V_{CC}}{2(R_{Ln} + R_{En})}$$
$$I_{C \max} = I_{CQ} (1 + \Delta I_{CQ})$$
$$I_{C \min} = I_{CQ} (1 - \Delta I_{CQ})$$

From these, we can calculate the base-emitter voltage under hot, highcurrent conditions (V_{BEX}) and under cold, low-current conditions (V_{BEN}) .

$$T_{\max} = \theta_{JA} I_{CQ} (V_{CC} / 2) + T_{A \max}$$

$$V_{BEX} = V_{BE1\min} + \Delta V_{BE} \log (I_{C\max}/I_1) - 0.0022 (T_{\max} - 25^{\circ}C)$$
$$T_{\min} = \theta_{JA} I_{CQ} (V_{CC}/2) (1 - (\Delta I_{CQ})^2) + T_{A\min}$$

$$V_{BEN} = V_{BE1 \max} + \Delta V_{BE} \log(I_{C \min}/I_1) - 0.0022(T_{\min} - 25^{\circ}C)$$

Now, a better estimate of the emitter resistance can be made:

$$R_{E(n+1)} = \frac{-2(V_{BEX} - V_{BEN})}{I_{C \max} - I_{C \min}}$$

From this point, if $V_{BEX} > V_{BEN}$, then R_E is set to zero, R_L is increased by 10% and the design procedure is repeated. Iterations continue until

$$\frac{R_{E(n+1)} - R_{En}}{R_{En}} < 0.5\%$$

If at any time the condition $T_{\text{max}} > T_J_{\text{max}}$ occurs, R_L is increased by 10%.

When the iterative procedure is complete, T_{max} , $I_{C \text{max}}$, T_{min} , and $I_{C \text{min}}$ are displayed.

124 5: Transistor Amplifier Bias Optimization

Values for

 $h_{FE \max}$ = maximum worst-case current gain at T_{\max} or T_{\min} and $I_{C \max}$ or $I_{C \min}$

and

 $h_{FE \min}$ = minimum worst-case current gain at T_{\max} or T_{\min} and $I_{C \max}$ or $I_{C \min}$

are determined from the transistor's data sheet. The Thevenin-equivalent resistance (R_B) and voltage (V_{BB}) of the amplifier's bias network are calculated:

$$R_{B} = \frac{h_{FE \max} h_{FE \min} \left[R_{E(n+1)} \left(I_{C \max} - I_{C \min} \right) + V_{BEX} - V_{BEN} \right]}{h_{FE \max} I_{C \min} - h_{FE \min} I_{C \max}}$$

 $V_{BB} = V_{BEN} + I_{C\min}((R_B / h_{FE\min}) + R_{E(n+1)})$

Now the bias resistors are calculated:

$$R_1 = \frac{R_B V_{CC}}{V_{BB}}$$

$$R_2 = \frac{R_B V_{CC}}{V_{CC} - V_{BB}}$$

Finally, the minimum power gain and minimum signal power are calculated:

$$A_{P} = \frac{R_{B} R_{L} h_{FE \min}}{R_{E} (R_{B} + h_{FE \min} R_{E})}$$
$$P_{S} = (1 - \Delta I_{CQ})^{2} \left(\frac{V_{CC}^{2} R_{L}}{8 (R_{L} + R_{E})^{2}} \right)$$

Variables Used.

In Equations	Description	In Program
V _{CC}	Source voltage (volts).	VCC
ΔI_{CQ}	Maximum desired percentage variation of quiescent current.	dIC0
T _{A max}	Maximum ambient temperature (use the maximum case temperature for a transistor mounted on a heat sink).	TAmax
$T_{A \min}$	Minimum ambient temperature.	TAmin
$T_{J \max}$	Maximum junction temperature.	TJmax
P _D	Maximum rated power dissipation at 25°C.	PD
I ₁	Collector current, usually selected for convenience so that I_1 and 10 I_1 at 25°C bracket the expected operating point.	I1
ΔV_{BE}	Typical base-emitter voltage change over the range of I_1 to 10 I_1 at 25°C.	dVBE
$V_{BE1 m min}$	Minimum base-emitter voltage at I_1 at 25°C.	VBE1min
$V_{BE1 \max}$	Maximum base-emitter voltage at I_1 at 25°C.	VBE1max
$h_{FE \max}$	Maximum worst-case current gain at T_{max} or T_{min} and $I_{C max}$ or $I_{C min}$.	R ₀₁
$h_{FE \min}$	$\begin{array}{l} \mbox{Minimum worst-case current} \\ \mbox{gain at } T_{max} \mbox{ or } T_{min} \mbox{ and } I_{\mathcal{C}max} \mbox{ or } \\ I_{\mathcal{C}min} \mbox{ .} \end{array}$	R ₀₂

Remarks.

- Flag 01 is used for branching control.
- Flag 02 is used to control the labels given to results.
- Flag 21 (printer enable) and flag 55 (printer existence) control printer output.
- Registers R₀₀ through R₀₉ are used for storing intermediate results. Be sure to set the SIZE to at least 10 registers (MODES) V SIZE 10 ENTER) before running "BIAS".

Program Instructions.

- 1. Key the "BIAS" program (listed on page 130) into your calculator.
- **2.** Press **XEQ BIRS** (to run the "BIAS" program).
- 3. Input the variables as prompted; press **R/S** after each entry.
- **4.** After the last input, the program calculates and displays T_{max} and I_{max} . The program then calculates and displays T_{min} and I_{min} .
- 5. When you see Hmax?, key in a value for the maximum worst-case current gain at T_{max} or T_{min} and I_{Cmax} or I_{Cmin} . Press [R/S].
- **6.** When you see Hmin?, key in a value for the minimum worst-case current gain at T_{max} or T_{min} and I_{Cmax} or I_{Cmin} . Press **R/S**.
- 7. The program then calculates values for R_E , R_L , R_1 , R_2 , and A_P . If you are not using a printer, press **R/S** after each result is displayed.

Example. A single-stage class "A" amplifier is connected to a 30-volt power supply. Calculate the maximum power output and maximum power gain obtained from a transistor over an ambient temperature range of 0°C to 70°C, with a maximum quiescent-current variation of $\pm 20\%$ (or .2).

From the transistor's data sheet,

 $TJ_{max} = 150^{\circ}C$ PD = 0.36 W $\Delta V_{BE} = 0.10 v \text{ from 3 to 30 mA}$ $V_{BE \, lmin} = 0.52 v \text{ at 3 mA at } 25^{\circ}C$ $V_{BE \, lmax} = 0.72 v \text{ at 3 mA at } 25^{\circ}C$

$I_1 = 0.001 \text{ A}$ $h_{FEmax} = 600$ $h_{FEmin} = 100$

Select ENG 2 display format and run the "BIAS" program.

DISP ENG 02 (XEQ) BIAS	Y: 0.00E0 VCC?0.00E0
30 (<u>R/S</u>)	Y: 30.0E0 dicq?0.00E0
.2 (R/S)	Y: 200.E-3 TAmax?0.00E0
70 (<u>R/S</u>)	Y: 70.0E0 TAmin?0.00E0
0 (<u>R/S</u>)	Y: 0.00E0 TJmax?0.00E0
150 (<u>R/S</u>)	Y: 150.E0 PD?0.00E0
.36 R/S	Y: 360.E-3 I1?0.00E0
.001 (R/S)	Y: 1.00E-3 dVBE?0.00E0

.1 (R/S)	Y:100.⊑-3 YBE1min?0.00⊑0
.52 (R/S)	Y: 520.E-3 VBE1max?0.00E0
.72 (<u>R/S</u>)	Тмах=148.Е0 Імах=18.0Е-З
(R/S)	Tmin=74.8E0 Imin=12.0E-3
(R/S)	Hmax? x: 500.e-3
Refer again to the transistor's data sheet a	nd input h _{FEmax} .
600 <u>R/S</u>	Hmin? x:600.⊑0
Now, input h _{FEmin} .	
100 (<u>R/S</u>)	RE= x: 115.¤0
(R/S)	RL= x: 888.e0
<u>[R/S]</u>	R2= x: 4.18E3
[R/S]	R1= x: 45.0E3

R/S

AP= x: 22.9⊑0

"BIAS" Program Listing.

Program:

Comments:

00 (553-Byte Prom) 01 LBL "BIAS" 02 INPUT "VCC" 03 INPUT "dICQ" 04 INPUT "TAmax" 05 STO 07 06 INPUT "TAmin" 07 STO 08 08 INPUT "TJmax" 09 INPUT "PD" 10 STO 09 11 INPUT "I1" 12 INPUT "dVBE" 13 INPUT "VBE1min" 14 INPUT "VBE1max" 15 CF 01 16 SF 02 17 SF 21 18 RCL "TJmax" 19 25 20 -21 RCL÷ 09 22 STO 09 23 RCL "VCC" 24 X+2 25 X 26 RCL "TJmax" 27 RCL- "TAmax" 28 4.4 29 X 30 ÷

Inputs values.

Initializes values for iterative process.

31 STO 03 32 0.1 33 × 34 STO 04 35 LBL 00 36 RCL "VCC" 37 2 38 ÷ 39 ENTER 40 ENTER 41 RCL 03 42 RCL+ 04 43 ÷ 44 STO 00 45 RCL× 09 46 × 47 RCL+ 07 48 RCL "TJmax" 49 X<>Y 50 X>Y? 51 GTO 05 52 XEQ 03 53 +/-54 RCL "dICQ" 55 1 56 + 57 XEQ 04 58 RCL+ "VBE1min" 59 STO 05 60 1 61 RCL "dICQ" 62 X+2 63 -64 2 65 ÷ 66 RCL× 00 67 RCL× 09 68 RCL× "VCC" 69 RCL+ 08 70 XEQ 03

Begins iterative loop.

```
71 +/-
72 1
73 RCL- "dICQ"
74 XEQ 04
75 RCL+ "VBE1max"
76 STO 06
77 RCL 05
78 X>Y?
79 GTO 02
80 -
81 RCL÷ 00
82 RCL÷ "dICQ"
83 RCL 04
84 X<>Y
85 STO 04
                              Repeats iterative loop as needed.
86 %CH
87 0.5
88 X<u>∠</u>Y?
89 GTO 00
90 FS? 01
91 GTO 01
92 SF 01
93 GTO 00
                              Prompts for h_{FEmax} and h_{FEmin}.
94 LBL 01
95 CF 01
96 "Hmax?"
97 PROMPT
98 STO 01
99 "Hmin?"
100 PROMPT
101 STO 02
102 X<>Y
                              Calculates R_B.
103 RCL "dICQ"
104 2
105 ×
106 RCL× 00
107 RCL× 04
108 RCL+ 05
```

109 RCL- 06 110 RCL× 02 111 RCL× 01 112 1 113 RCL- "dICQ" 114 RCL× 01 115 1 116 RCL+ "dICQ" 117 RCL× 02 118 -119 ÷ 120 RCL÷ 00 121 STO 07 Calculates V_{BB} . 122 RCL÷ 02 123 RCL+ 04 124 RCL× 00 125 1 126 RCL- "dICQ" 127 × 128 RCL+ 06 129 STO 08 Calculates R_1 . 130 RCL "VCC" 131 X<>Y 132 ÷ 133 RCL× 07 134 RCL "VCC" Calculates R_2 . 135 RCL× 07 136 LASTX 137 RCL- 08 138 ÷ Recalls R_L and R_E , and then 139 RCL 03 displays each of the four values in 140 RCL 04 the stack. 141 "RE" 142 XEQ 02 143 "RL" 144 XEQ 02 145 "R2" 146 XEQ 02

```
147 "R1"
148 XEQ 02
149 ÷
                               Calculates and then displays the
150 RCL× 02
                               power gain. (When the RTN at line
                               168 is reached, the program ends
151 RCL 07
152 ×
                               because there are no pending
                               subroutine calls.)
153 LASTX
154 RCL 04
155 RCL× 02
156 +
157 ÷
158 LOG
159 10
160 ×
161 "AP"
                               Displays a result and rolls the stack
162 LBL 02
163 ⊢"="
                               down one register (for the next
164 FS? 55
                               result).
165 ARCL ST X
166 AVIEW
167 R+
168 RTN
169 LBL 03
                               Calculates temperature.
170 FS? 01
171 XEQ A
172 25
173 -
174 2.25-3
175 ×
176 RTN
                               Displays proper label and
177 LBL A
178 "Tmax="
                               temperature.
179 FC? 02
180 "Tmin="
181 ARCL ST X
182 ⊢"'י
183 RTN
```

Calculates current. 184 LBL 04 185 RCL× 00 186 FS? 01 187 XEQ I 188 RCL÷ "I1" 189 LOG 190 RCL× "dVBE" 191 + 192 RTN Displays proper label and current. 193 LBL I 194 FS? 02 195 ⊢"Imax=" 196 FC?C 02 197 ⊢"Imin=" 198 ARCL ST X 199 AVIEW 200 RTN 201 STO 07 202 CLX 203 STO 04 Increases R_L by 10%. 204 LBL 05 205 1.1 206 STOX 03 207 GTO 00 208 END

Truth Tables

This chapter contains two programs for testing logical expressions. The first program, "PTTBL" (*print truth table*), allows you to print a complete truth table. The second program, "ITTBL" (*interactive truth table*), allows you to display any row of a truth table.

Writing a Logical Expression as a Program

Before using "PTTBL" or "ITTBL" you must enter a logical expression. This is done by writing a program that represents the expression.

Each of the storage registers R_{00} through R_{05} holds a 1 or 0. These are the inputs to the function that you write. In your program recall each register as it's needed and use the Boolean logic functions (AND, OR, and XOR) to create the expression.

Here's a simple example. This program represents the expression A OR B, where A is stored in R_{01} and B is stored in R_{00} .

```
00 ( 10-Byte Prom )
01 LBL "OR"
02 RCL 01
03 RCL 00
04 OR
05 END
```

The calculator's built-in NOT function returns the *36-bit* logical NOT of the number in the X-register. To perform a *single-bit* logical NOT, execute these three functions:

SIGN LASTX -

The logical expression

(A AND B AND C) OR (A AND \overline{B} AND \overline{C})

can be represented with the following program (assuming A is in R_{02} , B is in R_{01} , and C is in R_{00}).

"EXMPL" Program Listing.

Pro	gram:	Comments:
00 01 02 03 04 05 06	(31-Byte Prgm) LBL "EXMPL" RCL 02 RCL 01 AND RCL 00 AND	Calculates <i>A</i> AND <i>B</i> AND <i>C</i> .
07 08 09 10	RCL 00 SIGN LASTX -	Calculates \overline{C} .
11 12 13 14	RCL 01 SIGN LASTX -	Calculates \overline{B} .
15	AND	Calculates \overline{B} AND \overline{C} .
16 17	RCL 02 AND	Calculates A AND \overline{B} AND \overline{C} .
18 19	OR END	Calculates ($A \text{ AND } B \text{ AND } C$) OR ($A \text{ AND } \overline{B} \text{ AND } \overline{C}$)



Since the name of your program is stored in a variable (FCN), do not use a global label longer than six characters.

Printing a Truth Table ("PTTBL")

This program prints a truth table for a logical expression written as previously described. You must provide the name of the function (global label) and the number of the most significant input bit. (Bits are numbered right to left; the right-most bit is number 0 and is stored in R_{00} .)

Required Programs. "PTTBL" (page 141) and "FCN?" (page 156).

Variables Used.

Description	In Program
Function name.	FCN
Loop counter.	count
Most significant bit.	msb

Remarks.

- Registers R₀₀ through R₀₅ are used as input registers (bits) for the Boolean expression. Be sure to set the SIZE to at least six registers (MODES) SIZE 6 ENTER) before running "PTTBL".
- This program clears all of the storage registers.
- Flag 12 is used to produce double-wide output from the printer.

Program Instructions.

- 1. Key the "PTTBL" and "FCN?" programs into your calculator.
- 2. Key in the program that represents the logical expression (described in the "Writing a Logical Expression as a Program" section on page 137).
- **3.** Press **XEQ PTTBL** (to run the "PTTBL" program).
- 4. When you see Function Name?, use the ALPHA menu to type the name of the function (global label). Press **R/S**.

- 5. The program then prompts you for the most significant bit (msb?). Key in the number of the highest register you want to use as an input to your function $(1 \le msb \le 5)$.
- 6. Press **R/S** to print the truth table.

Example. Print a truth table for the expression on page 137. (If you haven't done it already, key in the "EXMPL" program on page 138.)

Run the "PTTBL" program.

XEQ PTTBL

Function Name? Maccal agen usual Norse (Sarow 1987)

Key in the name of the logical expression you want to print.

EXMPL

EXMPL_ MSCO3 FGHI JKLMI NOPQ RSTUW WXY2

R/S

Since the logical expression uses three inputs (A, B, and C), the most significant bit is stored in R_{02} .

Y:0 M≲b?0

2 **R/S**

Y: X:	3	
		_

Printer Output.

2	1	0	EXMPL
00001111	00110011	01010101	0000-001

"PTTBL" Program Listing.

00 (168-Byte Prom)

01 LBL "PTTBL"

07 INPUT "msb"

02 XEQ "FCN?"

Program:

03 PRON

08 1 09 +

04 CF 12 05 CLST 06 ALL

Comments:

Prompts for a function name using the "FCN?" utility. Printing is enabled.

Prompts for the number of the most significant bit and sets up the loop counter.

16 17	÷ STO "count"
18 19	CLA PRA
20	RCL "msb"
21	LBL 00
22	ARCL ST X
23	⊢" "
24	DSE ST X
25	GTO 00
26	ARCL ST X
27	⊢" "
28	ARCL "FCN"
29	⊢"';"
30	PRA
31	SF 12

Prints a single blank line.

Prints a table header. (Note that line 27 has seven blank spaces between the double quotes.)

32 LBL 01 Initializes the inputs (storage registers) and prepares to print a 33 CLRG row in the truth table. 34 RCL "count" 35 IP 36 RCL "msb" 37 LBL 02 Stores the input bits into the appropriate storage registers. 38 BIT? 39 XEQ 04 40 DSE ST X 41 GTO 02 42 BIT? 43 XEQ 04 44 XEQ IND "FCN" Evaluates the expression for the given inputs. 45 CLA Accumulates the input bits into the 46 RCL "msb" Alpha register. 47 LBL 03 48 ARCL IND ST X 49 DSE ST X 50 GTO 03 51 ARCL IND ST X Accumulates the output bit into the 52 F" Alpha register. (Note that line 52 53 ARCL ST Y has four blank spaces between the double quotes.) 54 PRA Prints a line in the truth table and 55 ISG "count" completes the loop. 56 GTO 01 57 CF 12 Resets the double-wide flag and 58 RTN ends. 59 LBL 04 Stores an input bit into the storage register identified in the Y-register. 60 1 61 STO IND ST Y 62 R+ 63 END
An Interactive Truth Table ("ITTBL")

This program allows you to change any of the inputs and display the corresponding output. Here's a typical display:

Given these inputs, the expression produces this output



Required Programs. "ITTBL" (page 145) and "FCN?" (page 156).

Remarks.

- FCN (the function name) is the only variable used by this program.
- Registers R₀₀ through R₀₅ are used as input registers (bits) for the Boolean expression. Be sure to set the SIZE to at least six registers (MODES) SIZE 6 ENTER) before running "ITTBL".
- This program clears all of the storage registers.
- This program does not produce printed output.

Program Instructions.

- 1. Key the "ITTBL" and "FCN?" programs into your calculator.
- 2. Key in the program that represents the logical expression (described in the "Writing a Logical Expression as a Program" section on page 137).
- **3.** Press **XEQ** ITTEL (to run the "ITTEL" program).
- 4. When you see Function Name?, use the ALPHA menu to type the name of the function (global label). Press **R/S**.
- 5. The program enters an interactive mode. The menu labels represent the inputs (R_{05} through R_{00}) as shown above.

To change one of the inputs, press the corresponding menu key. The program returns the result to the X-register. By toggling the inputs you can view the value of the expression for any combination of inputs.

6. To quit, press EXIT.

Example. Change the inputs and view the value of the expression. (If you haven't done it already, key in the "EXMPL" program on page 138.)

Run the "ITTBL" program.

XEQ ITTBL

Function Name? [13003.6400.0343][COSX (Sanow)[28934]

Key in the name of the logical expression you want to print. (If you just worked the example in the previous section, FCN probably still contains "EXMPL". Press •; if you see EXMPL, then simply press **R/S** to continue.)

EXMPL

EXMPL_ Miscos Fight Usumi Norr (Seturi Usiya)

x: 0					
Û	Û	Ú	Û	Û	Û

Toggle the most significant bit (bit 2 for this example) by pressing the fourth menu key.

0 (the LOG key)

Notice that the output (X-register) is now 1. Now toggle bits 1 and 0 (the fifth and sixth menu keys).

c: 1

0 (the LN key)

x: 0					
Ú	Ú.	Û	1	1	0

0 0 1 0 0

(the XEQ key)

V. 1					
∧ · ↓					
Ú	0	Ú	1	1	1
	-				

R/S

When you're finished, press **EXIT** to quit.

EXIT

Y: 1 X: 1

"ITTBL" Program Listing.

Program:

00 (125-Byte Pram) 01 LBL "ITTBL" 02 XEQ "FCN?" 03 ALENG 04 X=0? 05 GTO 09 06 BINM 07 ALL 08 CLRG 09 CLMENU 10 KEY 9 GTO 09 11 LBL A 12 CLA 13 ARCL 05 14 KEY 1 XEQ 01 15 CLA 16 ARCL 04 17 KEY 2 XEQ 02 18 CLA 19 ARCL 03 20 KEY 3 XEQ 03 21 CLA 22 ARCL 02 23 KEY 4 XEQ 04 24 CLA 25 ARCL 01 26 KEY 5 XEQ 05 27 CLA 28 ARCL 00 29 KEY 6 XEQ 06

Comments:

Prompts for a function name using the "FCN?" utility.

Initializes by selecting Binary mode, selecting ALL display format, and clearing the storage registers and programmable menu definitions.

Defines the **EXIT** key to branch to LBL 09. Defines the six top-row menu keys using the numbers in the corresponding storage registers as menu labels.

30	XEQ	IND	"FC	:N"	Evaluates the logical expression and leaves the value in the X-register.
31 32 33	MENU STOF GTO	J S A			Displays the menu and stops. Pressing R/S redisplays the menu.
34 35 36 37 38 40 41 42 44 45 46 47 48	LBL 5 GTO LBL 4 GTO LBL 3 GTO LBL 2 GTO LBL 1 GTO	01 07 02 03 07 04 07 05 07			Produces the appropriate register number, depending on which menu key is pressed.
49 50	LBL CLX	06			
51 52 53 54 55 56 57	LBL RCL SIGH LAS ^T - STO RTN	07 IND N TX IND	ST ST	× Y	Toggles the bit in a particular storage register.
58 59 60	LBL EXI END	09 TALL			Exits all menus and ends.

7

Utilities

The programs in this chapter are general-purpose utilities and subroutines used by other programs in this book. You may also find them useful when writing your own programs.

Circuit Calculation Utilities

Impedance of an Element (" $C \rightarrow Z$ " and " $L \rightarrow Z$ ")

This program converts the value for a capacitor or inductor in the X-register to a complex impedance. Before executing "C \rightarrow Z" (*capacitance to impedance*) or "L \rightarrow Z" (*inductance to impedance*), store the radian frequency in the variable w.

Remarks.

- w is the radian frequency, 2πf radians/second. (A lowercase "W" is used because the HP-42S does not have a lowercase omega character.)
- This program sets Rectangular mode.

"C \rightarrow Z" and "L \rightarrow Z" Program Listing.

Program:

Comments:

00	(30-Bute Pram)	
01 02	LBL "C→Z" XEQ 00	Calculates the impedance for the given capacitance (which is the
ØЗ	1/X	reciprocal of impedance for
04	RTN	inductance).
05	LBL "L+Z"	Calculates the impedance for the
06	LBL 00	given inductance.
07	RECT	-
08	RCL× "w"	
09	0	
10	X<>Y	
11	COMPLEX	
12	END	

Combining Parallel Impedances ("ZP")

The "ZP" program takes two complex impedances (in the X- and Y-registers) and returns the combined impedance for the two elements connected in parallel.

"ZP" Program Listing.

T
+
1/X
END

Entering Radian Frequency ("FQ?")

The "FQ?" program prompts for a value of w, the radian frequency value used by several programs in this book. Whenever you see Radian Frequency $(2\pi f)$?, key in the frequency and press **B/S**.

Remarks.

- w is the radian frequency, 2πf radians/second. (A lowercase "W" is used because the HP-42S does not have a lowercase omega character.)
- The program sets flag 25 (error ignore) to prevent an error from stopping the program if w doesn't exist.

"FQ?" Program Listing.

00	(44-Byte Prgm)	Ø6	RCL "w"
01	LBL "FQ?"	07	CF 25
02	"Radian Frequenc"	Ø8	PROMPT
03	H"y(2πf)?"	09	STO "w"
04	SF 25	10	END
05	CLX		

Circuit Element Input Utility ("EL?")

This program displays a menu for entering six types of circuit elements. It is designed to be used by other circuit analysis programs, such as the "CIRCT" program on page 45.

Each routine in this program displays a menu for entering an element using common units. For example, when you enter a resistor, the calculator displays:

> Value? OHM (KOHM | MOHM)

You can enter a 2,000-ohm resistor by pressing 2000 OHM, 2 KOHM, or .002 MOHM.

The program returns two numbers to the stack:

Y: ee.yyxxX: element value

where the X-register contains the *element value* you keyed in adjusted to the default units shown in the following table. *ee* is the element type in the table, and *yy* and *xx* are the location numbers entered into the stack. These numbers are used by programs such as "CIRCT" to indicate where a particular element occurs in a circuit.

Element Type	Units	Type Code		
Resistor	Ohms	82		
Capacitor	Farads	67		
Inductor	Henrys	76		
General impedance	Ohms	90*		
Voltage source	Volts	86*		
Current source	Amperes	73*		
*If the element value is a complex number, the type code is negative.				

Remarks.

- The type code is temporarily stored in a variable named *TYPE*.
- Flag 08 is set to indicate when an element has been entered successfully.

"EL?" Program Listing.

Program:

```
00 ( 374-Byte Pram )
01 LBL "EL?"
02 LBL A
03 RECT
04 CF 08
05 "R"
06 KEY 1 GTO 09
07 "C"
08 KEY 2 GTO 12
09 "L"
10 KEY 3 GTO 13
11 "Z"
12 KEY 4 GTO 10
13 "V"
14 KEY 5 GTO 14
15 "I"
16 KEY 6 GTO 15
17 KEY 9 GTO 99
```

Comments:

Defines menu for entering location and type of element. The **EXIT** key is defined to cause a branch to the "END" (which causes a return to the calling program). 18 "Location: " 19 H"# [ENTER] #" 20 MENU 21 PROMPT 22 GTO A 23 LBL 09 24 82 25 GTO 11 26 LBL 10 27 90 28 LBL 11 29 XEQ 00 30 "OHM" 31 KEY 1 XEQ 01 32 "KOHM" 33 KEY 2 XEQ 02 34 "MOHM" 35 KEY 3 XEQ 03 36 "S" 37 KEY 6 XEQ 08 38 GTO B 39 LBL 12 40 67 41 XEQ 00 42 "pF" 43 KEY 1 XEQ 07 44 "nF" 45 KEY 2 XEQ 06 "Fע" 46 47 KEY 3 XEQ 05 48 "mF" 49 KEY 4 XEQ 04 50 "E" 51 KEY 5 XEQ 01 52 GTO B

Displays the input message and the menu. Pressing $\boxed{R/S}$ redisplays the menu.

Enters the code for a resistor.

Enters the code for a general impedance.

Defines a menu for entering units for a resistor of general impedance. (The \Im in line 36 refers to the SI unit "siemens.")

Enters the code for a capacitor and defines a menu for entering units of capacitance.

53 LBL 13 Enters the code for an inductor and 54 76 defines a menu for entering units of 55 XEQ 00 inductance. 56 "mH" 57 KEY 1 XEQ 04 58 "H" 59 KEY 2 XEQ 01 60 "KH" 61 KEY 3 XEQ 02 62 "MH" 63 KEY 4 XEQ 03 64 GTO B 65 LBL 14 Enters the code for a voltage source 66 POLAR and defines a menu for entering 67 86 voltage units. 68 XEQ 00 69 "mV" 70 KEY 1 XEQ 04 71 "V" 72 KEY 2 XEQ 01 73 "KV" 74 KEY 3 XEQ 02 75 GTO B 76 LBL 15 Enters the code for a current source 77 POLAR and defines a menu for entering 78 73 current units. 79 XEQ 00 80 "mA" 81 KEY 1 XEQ 04 82 "A" 83 KEY 2 XEQ 01 84 "KA" 85 KEY 3 XEQ 02 86 GTO B

87 LBL 08 88 17X 89 RTN 90 LBL 07 91 XEQ 04 92 LBL 06 93 XEQ 04 94 LBL 05 95 XEQ 04 96 LBL 04 97 1E3 98 ÷ 99 RTN 100 LBL 03 101 XEQ 02 102 LBL 02 103 163 104 × 105 LBL 01 106 RTN 107 LBL 00 108 X<>Y 109 1 110 % 111 R+ 112 + 113 1 114 % 115 R+ 116 + 117 STO "TYPE" 118 CLMENU 119 KEY 9 GTO A 120 RTN

Adjusts the element value to be expressed in the *default* units.

Combines the location number and element type code into a single number and stores it in the variable *TYPE*.

Clears the programmable menu and defines the **EXIT** key to return to the first menu.

121	LBL B
122	MENU
123	1
124	"Value?"
125	PROMPT
126	SF 08
127	RCL "TYPE"
128	CLV "TYPE"
129	X<>Y
130	REAL?
131	RTN
132	X<>Y
133	+/-
134	X<>Y
135	LBL 99
136	END

Displays the units menu and prompts for a value. Pressing $[\overline{R/S}]$ causes the default value of 1 to be used.

Sets flag 08 to indicate an element has been entered. Returns the type code in the Y-register and the element value in the X-register. If the element value is complex, the type code is made negative.

Ends the program.

Function Name Utility ("FCN?")

This program prompts for a function name and then stores the name into a variable named FCN. If FCN contains a string, it is recalled into the Alpha register (press • to clear the Function Name? message). Use the ALPHA menu to type the name of the function (global program label) and then press \mathbb{R}/\mathbb{S} to continue.

Remarks.

Program:

- FCN contains the variable name (up to six characters).
- Flag 21 (printer enable) is cleared by the program.
- Flag 25 (error ignore) is used to prevent the program from stopping if FCN doesn't exist.
- The plotting programs in the owner's manual use a similar technique to prompt for a function name. If you have the "FCN?" program in your calculator, you can shorten one or both of the plotting programs by calling "FCN?" as a subroutine (XEQ "FCN?").

"FCN?" Program Listing.

Comments:

LBL "FCN?" "Function Name?" CF 21 AVIEW	Displays the prompt.
SF 25 RCL "FCN" CF 25 CLA STR? ARCL ST X	If FCN exists and contains an Alpha string, recalls that string to the Alpha register.
AON STOP	Turns on the ALPHA menu and stops.
AOFF ASTO "FCN" END	Turns off the ALPHA menu and stores the first six characters in the Alpha menu into <i>FCN</i> .
	LBL "FCN?" "Function Name?" CF 21 AVIEW SF 25 RCL "FCN" CF 25 CLA STR? ARCL ST X AON STOP AOFF ASTO "FCN" END

Yes/No Utility ("Y?N")

This program displays a menu for a Yes/No decision. It returns a zero if you press NO, R/S, or EXIT; it returns a 1 if you press YES.

If you want to use this utility in your own programs, simply place a message in the Alpha register and then execute the utility (XEQ "Y?N"). Your program can then test the X-register to detect a "yes" (1) or a "no" (0).

Remarks.

- Flag 21 (printer enable) is cleared by this program.
- This program redefines the programmable menu.

"Y?N" Program Listing.

00	(41-Byte Prom)	09 KEY 9 GTO 00
01	LBL "Y?N"	10 MENU
02	CF 21	11 STOP
03	AVIEW	12 LBL 00
04	CLMENU	13 0
05	"YES"	14 RTN
06	KEY 1 GTO 01	15 LBL 01
07	"NO"	16 1
08	KEY 6 GTO 00	17 END

Product Over Sum Utility ("P/S")

This routine is quite useful for many electrical engineering applications. It simply takes two values (in the X- and Y-registers) and returns the product of the two values divided by the sum of the two values:

 $\frac{xy}{x+y}$



If x + y = 0, the program will error at line 05 (Divide by \emptyset).

```
"P/S" Program Listing.
```

00 (15-Byte Prgm) 01 LBL "P/S" 02 RCL× ST Y 03 X<>Y 04 RCL+ ST L 05 ÷ 06 END

Size Utility ("SIZE?")

This program returns the number of storage registers available.

"SIZE?" Program Listing.

00	{ 27-Byte Prom	>	06	FC?C	25
01	LBL "SIZE?"		07	RTN	
02	SF 25		Ø8	DIM?	
03	RCL "REGS"		09	×	
04	FC? 25		10	END	
05	0				

Electrical Engineering contains a variety of programs and examples to provide solutions for electrical engineers and engineering students.

Circuit Calculations

Voltage Division • Current Division • Power Triangle • Frequency Response of Transfer Function • RC Timing • Delta-Wye Conversions

Network Analysis

Using the Circuit Editor • Mesh Analysis • Nodal Analysis • Impedance of a Ladder Network

Filter Design

Active Filter Design • Butterworth Filter Design

Transmission Lines

Transmission Line Calculations • Transmission Line Impedance

Amplifier Analysis

Transistor Amplifier Performance • Transistor Amplifier Bias Optimization

Truth Tables

Writing a Logical Expression as a Program • Printing a Truth Table • An Interactive Truth Table

Utilities

Circuit Calculation Utilities • Circuit Element Input Utility • Function Name Utility • Yes/No Utility • Product Over Sum Utility • Size Utility



Reorder Number 00042-90021

00042-90023 English Printed in Canada 11/88

